

SEGA[®] COMPUTER

The Official Sega User Club Magazine

MAY/JUNE 1986 ISSUE

In this Issue:
KARATE
THE CASE
WARI
MORE FORTH NOTES
3D MAZE
and Much Much more!



SEGA COMPUTER
the official magazine of the SEGA USER CLUB
New Zealand

Registered as a magazine at the CPO Wellington, New Zealand. Published bi-monthly. Magazine published about the middle of July, Sept, Nov, Jan, Mar, May.

ISSN

Published by:
Sega Software Support
P.O. Box 10-207
Auckland, 4.
NEW ZEALAND.

Telephone (09) 603-274

Editor: Glenys Millington

All contributions welcome. Name, address and phone number must be included.

All software programs published by the magazine become the property of Sega Software Support unless by prior arrangement. They are accepted on the basis that they are the original work of the author. The programs must be submitted on tape or disc and a printed listing is desirable. Software is not returned unless accompanied by a stamped self addressed envelope.

For overseas contributions, please enclose a \$1 note/coin of the country of origin if software is to be returned.

A plea for software listings — please check your software thoroughly for errors before sending it to us. Please update us on any errors you know about so that we can publish corrections.

We pay \$NZ20.00 for the feature software program published each month and \$NZ4.00 for all other full programs published.

Articles or reviews must be legible. Articles or reviews are paid \$NZ4.00 per page on a prorata basis.

All contributions are subject to approval by the editor and may be edited to suit the magazine style. All payments are made after publication.

SEGA USER CLUB MEMBERSHIP YEAR JULY — JUNE

Membership includes a subscription to the Club magazine SEGA COMPUTER as well as qualifying for other club benefits.

NEW ZEALAND

Membership Subscription: \$37.00 plus GST Total \$40.09

Back copies available: \$ 6.50 each
\$35.00 complete 1985/86 year

If you join after the start of the subscription year you may request the back issues for that year.

AUSTRALIA

Membership Subscription: \$NZ50.00 Airmail
\$NZ46.00 Surface Mail

ELSEWHERE

Membership Subscriptions available on request.

CONTENTS

Letters to Editor.....	2
Important Announcement.....	3
Some more FORTH notes.....	6
Wari.....	7
How to get HuCal to work.....	10
The Fish Tank.....	11
Code Breaker.....	12
The Case.....	15
SpaceFighter.....	20
Draw.....	22
Turtle.....	24
3D Maze.....	25
Piano Programme.....	29
Sega Users Club Contacts.....	32

EDITORIAL

Well, here we are again for another issue of the Sega mag, and what an issue it is! I would like to thank all those people who sent in programs to me to be placed inside these illustrious covers! I hope that you all keep it up, just remember, each time you send in a program the magazine is that bit closer to being published, so keep them flowing.

One thing that you will find of interest is the way the market is changing, a few years ago you could sell any machine quite easily, today things are a lot more difficult mainly because people want their computers to do three things. (1) Teach them about computers, (2) Do work for them and (3) Play games, in that order! A while ago, games playing was at the top of the list.

By the way, Sega Software Support are now taking over the magazine and the software for Sega. They're great people and you'll hear a lot more from them.

So without further ado, any more waffle, no back chat, meaningless monologue, or dreadful dribble... on with the mag!



Michael Howard
PROGRAMMER & TECHNICAL SUPPORT

LETTERS TO THE EDITOR



DEAR EDITOR

I have obtained a score of 999880 on Video Flipper and stopped to take a photo (enclosed) just to prove it. I don't receive your magazines but when a friend has finished with them, he gives them to me.

Andre Stokes, Huntly

Great stuff Andre! Unfortunately the photos were unreproducible in the mag. You really should get the mag because it makes up an important library of Sega information. This gives us a good idea . . . for mag subscribers we will start, in the next issue, a SCOREBOARD of hi-scores for popular games. Send us your high scores and we'll publish the best! Note that the next issue is the start of the new subscription year so you'll need to act fast!

DEAR EDITOR

Do you have any programmers manuals for the Sega BASIC?

Christopher Campbell, New Plymouth

C. E. Fenwick, Christchurch

This is a question that crops up constantly. The new catalogue from Sega Software Support has BASIC GAMES PROGRAMMING and other books which give a lot of help. Other than these books, the SEGA BASIC is virtually identical to Microsoft BASIC which has become an industry standard for the IBM PC and other business computers. Your local Whitcoulls Technical Books or University Bookshop will have books on Microsoft BASIC which you would find very helpful. In fact, one of the strengths of the Sega is the close compatibility in its BASIC to business computers. We intend publishing a full manual for the SEGA BASIC within the next 6 months.

DEAR EDITOR

I am interested in the LODE RUNNER cartridge. Is it still available? Can we expect any new titles? As a matter of interest, how are cartridges constructed and what advantages do they have over programs on disk? I assume they are mostly comprised of ROM.

When attempting to SORT using Hucal, can you please clarify what should happen with Sorting Keys 2 and 3 (Manual p. 34). I deleted my totals, experimented and found the results rather confusing.

R. J. Thorpe, Napier

Firstly, Lode Runner is unavailable as is Exerion, Safari Hunter and others. The only cartridges currently available are in the catalogue of Sega Software Support. However, don't give up hope . . . changes are afoot.

Cartridges are constructed from ROM chips containing the games program and a small logic circuit to enable it to be read by the computer. Their advantages are that they are very easy to use, require no extra hardware, instantly load and, of course, have good copy protection. Their disadvantages have always been cost. If I had my way, I would have every good

bit of software on cartridge but the cost would be exceptionally high.

Can anyone out there using Hucal give us some help on his last question?

DEAR EDITOR

I visited your offices when I was in Auckland and discussed how services to Sega owners could be improved.

I have talked to several Sega owners and it is clear there is a need for the following:

1. There is a criticism of the explanatory literature that goes with the computer and with the software. It does not provide sufficient data to understand what is happening.
2. There is a need to reinstate the series "Unexplained Sega Commands" in the mag.
3. What is happening to the magazine?

Keith Hovell, Dunedin

Dear Keith, I agree on all points. See my answer to one other letter here about Sega BASIC. We will reinstate the series about Unexplained Sega Commands.

As to the magazine, see elsewhere in the mag as to what is happening. Thanks for the letter. Much appreciated.

DEAR EDITOR

We have only had 4 issues so far of the magazine. What has happened?

Mrs McDonald, Auckland

Dear Mrs McDonald, due to some problems that arose late last year, the magazine was held up for 5 months. A double issue was released recently and now this issue is in your hands — the May/June issue . . . a little later than expected.

This is the last of the current subscription year. Sega Software Support will be handling both the Sega User Club and the magazine as well as software and hardware. Grandstand Computers will continue servicing all Sega equipment. Sega Software Support is immediately starting the production of the first magazine of the new subscription year (July/August) so that the magazines can be on time in future.

(P.S. Send in your subscriptions, programs and articles . . . NOW!!)

DEAR EDITOR

I am writing to complain that there are no outlets anywhere for Sega software or hardware. What has happened?

R. Tennyson, Timaru

That's right, few shops now stock Sega products. The only people who do are Sega Software Support. They operate a mail order service throughout NZ. Ask for their free catalogue c/- P.O. Box 10-207, Auckland 4.

IMPORTANT ANNOUNCEMENT

Bringing you up to date with the world of **SEGA**.

As some Sega Users have found out already, Grandstand Computers have now passed the responsibility for Sega support over to Glenys Millington at Sega Software Support. She will be responsible for all software and hardware purchases and for the **SEGA USER CLUB** and **Sega Computer**, the official User Club magazine. Mike Howard and all the Sega geniuses at Grandstand are still behind us and you can be assured that the Club and Magazine will continue at its usual high standard. However, we will make some improvements due to popular demand and these are shown over the page.

SEGA SOFTWARE SUPPORT is now responsible for all Sega enquiries apart from servicing equipment — these should still be made direct to Grandstand.

You can reach us at P.O. Box 10-207, Auckland 4, or Phone (09) 603-274 during business hours only.

This is the **last** magazine of the current Club subscription year and the next magazine (the July/August issue) is the start of a new Club subscription year. The new subscription costs are below. Unfortunately, GST has to be charged! Not our fault — blame Mr Douglas. To receive the next magazine we must have your Club subscription as soon as possible!! Get your chequebooks out now!! (and pens of course). As current subscribers know, the savings they make on our special software and hardware offerings more than make up for the Club subscription cost. The retail price of the magazine alone is \$6.95.

Sega Software Support will supply you by mail order with masses of software and some hardware . . . all excellent value! You are strongly advised to **purchase before GST** as this will raise the cost to you by at least 10%. Also, some of our stocks are limited.

Subscriptions for the **SEGA USER CLUB** (which includes **SEGA COMPUTER** magazine) can be obtained by completing the Subscription Order Form. Note that the retail price of the magazine is \$6.95 per issue.

The annual club subscription is \$37.00 plus GST.

Subs received before August 31: \$3.09 GST.

Subs received after August 31: \$3.70 GST.

Note: This assumes GST is 10%. If the GST rate alters we reserve the right to amend these charges.

Australian Subscribers: Surface Mail \$46.00 Airmail \$50.00

FREE GIFT FOR SUBSCRIBERS — SEE OPPOSITE PAGE

Contributions from all you Sega users for **your** magazine are always gratefully accepted. Program listings etc., where possible please. Keep them rolling in!

Do you know of someone who has a Sega and isn't on our mailing list? Send us their name and address. They'll thank you for it.

Because we know our quality of service and support we offer we will give you our guarantee — your subscription will be refunded on a pro rata basis if you are not 100% satisfied with the Club.

Regards



Glenys Millington
Manager

NEXT ISSUE NEXT ISSUE NEXT ISSUE

Due out at the end of August
Copy deadline: in our hands by 12 August!!!

Some of the changes for the next issue are:

1. A Buy/Sell Column

We are going to introduce a BUY/SELL Column for Hardware and user-written Software. Just send us:

- ★ Your name, address and phone number.
- ★ a few lines of advertising copy.
- ★ \$4.00 per insertion. Cash or cheque must be enclosed.

We reserve the right to edit or reject any advertisement. We obviously cannot accept responsibility for goods or services purchased from these advertisements.

2. An Australian column or page

We now have a number of contacts across the Tasman, many of whom get our magazine. Any Australian Sega user can just send us their material and we will take it from there.

3. We now take Bankcard or Visa payments.

4. An updated Local User Group list.

Many of the local user groups have fallen by the wayside over the past year due to changes of people and other vagaries. If you would like to be the User Group contact in your area, please send us your **name, area and phone number** and we will publish it for you. Others in your area can contact you and between yourselves arrange suitable times and places.

**July 1986 — June 1987 Sega User Club Subscription
(includes Sega Computer Magazine)**

Please send me in my FREE GIFT of COMPUTERS DON'T BYTE as I am not charging my sub. to a Credit Card and the date is before August 25.

Name:

Address:

.....

.....

New Member Renewal Member (tick one)

Payment Enclosed: \$37.00 + \$3.09 GST (Total \$40.09)


Cash/Cheque/Visa/Bankcard (circle one)
Cheques payable to Sega Software Support, PO Box 10-207, Auckland 4, N.Z.

Credit Card Number:

Expiry Date: / / Signed:.....

Free Gift to Subscribers...

COMPUTERS DON'T BYTE



Yes — we are
offering this Book
if you subscribe to
the SEGA USER
CLUB

Before 25th
August 1986

COMPUTERS DON'T BYTE—
A gift for children and adults alike.

The computer revolution is here and affecting you and your children. After all, that's why you bought your Sega computer. This book delightfully and amusingly fills in the detail of the march of the revolution step by step. The history and development of computers and their languages, how computers are being used in education, business and society and where the future is going — it's all here in **Computers Don't Byte**.

Written simply and in plain language this book will appeal to new computer owners as well those advanced in computing skills.

Mary Matthew has been explaining computers for many years and the text is accompanied by delightful cartoons by Rita Parkinson.

This is a book for the whole family and is widely used in schools throughout the country. Now is your chance to get a copy for yourself — FREE.

Some more FORTH notes . . . Michael Howard

My FORTH emulator program created a lot of interest from ardent readers. It's nice to see people realising that there is more to computers than just BASIC and games!

Firstly, here is a program written by Grant Emms. The program is written entirely in FORTH, and is called Lunar Lander. Remember, as with all FORTH programs, beware of spaces, and make sure you put them in the right places.

```
: LUNAR TITLE SET MAIN ;

: MAIN PRINT FUZ? SPEC HIEC LAND? MAIN ;

: SET 100 200 RAND 5 8 RAND 100 5 ;

: PRINT CR .' Fuel : ' 2 PICK . CR .' Height : ' 4 PICK . CR .' Speed: '
  DUP . CR .' Gravity: ' 3 PICK . ;

: FUZ? 2 PICK 0 <> IF INFUEL FLI? SUBF THEN ;

: INFUEL CR .' Fuel to use? ' KEY 48 - ABS DUP DUP . 10 > IF INFUEL
  THEN DUP ;

: FLI? 4 PICK > IF DROP 2 PICK THEN ;

: SUBF 2 ROLLER DUP 6 ROLL ROT - SWAP 1 ROLLER ;

: SPEC 2* - 3 PICK + ;

: ROLLER 0 DO 5 ROLL SWAP LOOP ;

: HIEC 4 ROLL 2 PICK - 0 3 ROLLER DROP ;

: LAND? 4 PICK 1 < IF LANDING THEN ;

: LANDING DUP 5 < IF CR .' We have Touchdown! ' THEN CRASH? ;

: TITLE 12 EMIT CR .' <<Lunar Lander>> ' CR ;

: CRASH? DUP 5 > IF CR .' We have a bit of a crater! ' THEN ABORT ;
```

I haven't had time to test this program out, but Grant tells me it works! To run the program just type in LUNAR, when you have finished typing in the program.

Other interesting remarks that Grant came up with include a way of speeding up the program. In FORTH you are forever PUSH'ing and POP'ing, so Grant sped up this bit by using Machine Code, here is that code:

```
10 E=0
20 X=&HF000,RESTORE 50
30 READ A$;IF A$="END" THEN 60
40 POKE X,VAL("&H"+A$);X=X+1;GOTO 30
50 DATA ED,5B,64,81,13,13,21,00,FA,01,08,00,ED,B0,11,00,FA,01,98,00,ED,B0,C9,END
60 X=&HF100,RESTORE 90
70 READ A$;IF A$="END" THEN 250
80 POKE X,VAL("&H"+A$);X=X+1;GOTO 70
90 DATA 11,A7,FA,21,9F,FA,06,14,C5,00,00,01,08,00,ED,B8,C1,10,F5,54,5D,2A,64,
  81,23,23,13,01,08,00,ED,B0,C9,END
```

And alter lines 365 and 370 to:

```
365 FOR Q=1 TO 4;CALL &HF000;A(Q)=E;NEXT;FOR Q=4 TO 1 STEP -1;E=A(Q);CALL
&HF000;NEXT
370 FOR Q=1 TO 4;PRINT TAB(8);Q;",";A(Q);NEXT;GOTO 410
```

Apparently a bug got into the program, line 1220 should also check for e\$="<>" as well. . . . SORRY!
Insert OR e\$ "<>" into the line.

Well that wraps up this bit on FORTH, I hope that people send in more modifications to me and we'll end up with a really great program. Don't forget that I will be more than happy to receive your FORTH programs.

WARI by Norman A. Raynel

This is an old African native game that has been played for hundreds of years. You can play against the computer or against a friend. There are 12 piles of stones (6 are yours and 6 are your opponents). Each pile has 4 stones at the start but this will vary as the game progresses. The object of the game is to select a pile of stones, pick them up and then one at a time drop them off around the other piles of stones in an anticlockwise direction. The computer will do this for you, all you have to do is select which pile of stones. If your last stone dropped on your opponents side and it makes a total of 2 or 3 in that pile then you capture all those stones and these will be transferred to your score. In addition, if the next pile of any consecutive pile in a clockwise direction has 2 or 3 stones in them then you will also capture those stones and they will be transferred to your score. The first player to gain 24 stones is the winner. If during the course of the game one player has no stones left then his opponent has to select a move that will give at least one stone.

Sounds complicated? It's not really, once you get used to it.

```
5 SCREEN 1,1:CLS
10 PRINT " WARI by Norman Raynel"
20 PRINT:PRINT :PRINT
30 PRINT"Welcome to the game of WARI."
40 PRINT "This is a very old native game that has been played for hundreds of ye
ars."
50 PRINT "In this game you have 6 piles of stones. Each pile has 4 stones at the
start but this will change as the game progresses."
55 PRINT "You can play against another human or against the computer. Either way
you will be PLAYER 2. on the red portion of the screen."
56 INPUT "Press CR to continue";D$
57 IF D$=CHR$(32) THEN CONT
60 PRINT "To play you must pick up all the stones from one pile of stones on you
r side and drop them off one at a time around the screen anti-clockwise."
65 PRINT "The computer will do this for you,all you have to do is select the pil
e number,1 to 6 for player 2 and 7 to 12 for player 1"
70 PRINT "The object of the game is to drop your last stone on your opponents si
de so that you make that pile 2 or 3 stones."
75 PRINT "If you do then you will capture all those stones plus any consecutive
piles clockwise around your opponents side that has 2 or 3 stones."
80 PRINT "The first player to capture 24 stones wins the game. If your opponent
has no stones on his side you must give him some on your next move."
85 PRINT "If you make a wrong move the computer will tell you that it is ILLEGAL
and you will be penalised one turn."
90 INPUT "Press CR to continue";D$
91 IF D$=CHR$(32) THEN CONT
100 REM "WARI"
110 J=1:K=1:Q=14:P=13:F=50:D=12
120 DIM T(Q),Y(Q),W(Q),V(6),E(6),B(Q)
130 ZB=RND(1):ZB=ZB/Q:ZA=.25+ZB:ZB=.25-ZB:GOSUB 780
140 FOR J=1 TO D:B(J)=4:NEXT :B(P)=0:B(Q)=0:MN=0:GOSUB 1100:GOSUB 910
150 SCREEN 2,2:GOSUB 1060:PRINT "WHO WANTS TO PLAY? PRESS C FOR COMPUTER, H FOR
HUMAN";G$=INKEY$
160 IF INKEY$=""THEN 160
170 IF INKEY$="H" THEN 1190
180 IF INKEY$="C" THEN 190
190 GOSUB 1060:PRINT "DO YOU WANT TO GO FIRST? Y/N "
200 IF INKEY$=""THEN 200
210 IF INKEY$="Y"THEN 290
220 GOSUB 1060:GOSUB 1070:PRINT "THINKING";:GOSUB 560
230 IF M<1 THEN 1240
240 GOSUB 1070:BEEP:PRINT "MY MOVE IS ";M
250 FOR J=1 TO Q:T(J)=B(J):NEXT :GOSUB 400
260 FOR J=1 TO Q:B(J)=T(J):NEXT :GOSUB 910
270 IF B(Q)<24 THEN 290
280 GOSUB 1070:PRINT "I WIN!";D$:GOTO 820
290 GOSUB 1060:BEEP:PRINT "YOUR MOVE? ";:R$=INKEY$
300 IF INKEY$="" THEN 300
310 R$=INKEY$:PRINT R$
320 M=INT(VAL(R$)):IF M>6ORM<1 THEN GOSUB 390
330 FOR J=1 TO Q:T(J)=B(J):NEXT
340 GOSUB 400:IF M<0 THEN GOSUB 390
350 FOR J=1 TO Q:B(J)=T(J):NEXT
360 MN=MN+1:GOSUB 910
370 IF B(P)<24 THEN 220
380 GOSUB 1070:BEEP2:PRINT "YOU WIN!";D$:GOTO 820
```



```

390 GOSUB 1060:PRINT TAB(15);" ILLEGAL ";:FOR J=1 TO 1000:NEXT :M=0:RETURN
400 IF T(M)=0 THEN M=-1:RETURN
410 R$="H":IF M>6 THEN R$="C":GOTO 430
420 FOR J=1 TO Q:Y(J)=T(J):NEXT :GOTO 450
430 FOR J=1 TO 6:Y(J)=T(J+6):Y(J+6)=T(J): NEXT
440 Y(P)=T(Q):Y(Q)=T(P):M=M-6
450 C=M:N=Y(C):FOR J=1 TO N:C=C+1
460 IF C=P THEN C=1
470 IF C=M THEN C=C+1:GOTO 460
480 Y(C)=Y(C)+1:NEXT :Y(M)=0:L=C
490 IF L<7OR Y(L)>3OR Y(L)<2 THEN 510
500 Y(P)=Y(P)+Y(L):Y(L)=0:L=L-1:GOTO 490
510 S=0:FOR J=7 TO D:S=S+Y(J):NEXT
520 IF S=0 THEN M=-2:RETURN
530 IF R$="H" THEN FOR J=1 TO Q:T(J)=Y(J):NEXT :RETURN
540 FOR J=1 TO 6:T(J)=Y(J+6):T(J+6)=Y(J):NEXT
550 T(Q)=Y(P):T(P)=Y(Q):RETURN
560 FOR A=1 TO 6:M=A+6:IF B(M)=0 THEN E(A)=-F:GOTO 750
570 FOR J=1 TO Q:T(J)=B(J):NEXT :GOSUB 400
580 IF M<0 THEN E(A)=-F:GOTO 750
590 IF T(Q)>23 THEN M=A+6:RETURN
600 GOTO 650
610 FOR J=1 TO Q:W(J)=T(J):NEXT :FOR K=1 TO 6
620 IF T(K)=0 THEN V(K)=F:GOTO 730
630 FOR J=1 TO Q:T(J)=W(J):NEXT :M=K:GOSUB 400
640 IF M<0 THEN V(K)=F:GOTO 730
650 FA=0:FB=.05:FC=0:FD=0:FOR J=7 TO D
660 FB=FB+T(J):IF T(J)>0 THEN FA=FA+1
670 IF T(J)<3 THEN FC=FC+1
680 IF T(J)>FD THEN FD=T(J)
690 NEXT :FE=FB:FOR J=1 TO 6:FE=FE+T(J):NEXT
700 FA=FA/6:FD=1-FD/FB:FC=1-FC/6:FB=FB/FE
710 V(K)=ZA*(FA+FB)+ZB*(FC+FD)+T(Q)+B(P)-B(Q)-T(P)
720 E(A)=V(K):GOTO 750
730 NEXT :E(A)=F:FOR J=1 TO 6:IF V(J)<E(A) THEN E(A)=V(J)
740 NEXT
750 PRINT ".":NEXT :M=0:FA=-F:FOR J=1 TO 6
760 IF (E(J)>FA) THEN FA=E(J):M=J+6
770 NEXT :RETURN
780 A$="-":FOR J=1 TO 23:A$=A$+"-":NEXT
790 C$=CHR$(95):FOR J=1 TO 23:C$=C$+CHR$(95):NEXT
800 D$=CHR$(8):FOR J=1 TO 5:D$=D$+D$:NEXT
810 RETURN
820 PRINT "GOOD GAME!"
830 BEEP:PRINT "WANT TO PLAY AGAIN? Y/N":R$=INKEY$
840 IF INKEY$="" THEN 840
850 IF INKEY$="Y" THEN 130
860 IF INKEY$="N" THEN 890
870 R$=LEFT$(R$,1):IF R$="Y" THEN 130
880 IF R$<>"N" THEN 830
890 SCREEN 1,1:CLS:PRINT "SEE YOU LATER"
900 PRINT :END
910 CURSOR 1,60
920 FOR J=0 TO 5:PRINT TAB(4*J+1);CHR$(8);CHR$(8);CHR$(8);
930 NEXT
940 CURSOR 1,60
950 FOR J=0 TO 5:PRINT TAB(4*J+1);B(12-J);:IF B(12-J)=0 THEN GOSUB 1090
960 NEXT :CURSOR 160,60:PRINT CHR$(8);CHR$(8);CHR$(8);CHR$(8);CHR$(8);CHR$(8);CHR
$(8);CHR$(8);CHR$(8);CHR$(8);CHR$(8);CHR$(8);CHR$(8);
970 CURSOR 160,60:PRINT "PLAYER 1: ";B(Q)
980 CURSOR 1,85:FOR J=0 TO 5
990 PRINT TAB(4*J+1);CHR$(8);CHR$(8);CHR$(8);
1000 NEXT
1010 CURSOR 1,85
1020 FOR J=0 TO 5:PRINT TAB(4*J+1);B(J+1);:IF B(J+1)=0 THEN GOSUB 1090
1030 NEXT: CURSOR 185,85:PRINT CHR$(8);CHR$(8);CHR$(8);CHR$(8);CHR$(8);CHR$(8);CH
R$(8);CHR$(8);CHR$(8);CHR$(8);
1040 CURSOR 160,85:PRINT "PLAYER 2: ";B(P)
1050 RETURN
1060 CURSOR 8,140:PRINT CHR$(5):CURSOR 8,140:RETURN
1070 CURSOR 8,160:PRINT CHR$(5):CURSOR 8,160:RETURN
1080 SCREEN 1,1:CLS:CURSOR 4,20:RETURN

```

```

1090 PRINT CHR$(8);:RETURN
1100 SCREEN 2,2:CLS:COLOR1,11,(0,10)-(255,75),4:COLOR1,9,(0,75)-(255,125),4:COLOR1,15,(0,125)-(255,175),4:PRINT TAB(16);"WARI"
1110 PRINT TAB(16);"----"
1120 PRINT :PRINT TAB(8);"PLAYER 1":PRINT
1130 FOR J=0 TO 5:PRINT TAB(4*J+1);12-J;:NEXT
1140 PRINT TAB(30);"CAPTURED":PRINT C$;TAB(30);LEFT$(A$,8)
1150 PRINT :PRINT :PRINT :PRINT :PRINT :PRINT C$:PRINT
1160 FOR J=0 TO 5:PRINT TAB(4*J+1);J+1;:NEXT
1170 PRINT :PRINT :PRINT TAB(8);"PLAYER 2 "
1180 PRINT :RETURN
1190 GOSUB 1060:SCREEN 1,1:CLS:BEEP:PRINT "WHAT ARE YOUR NAMES?";:INPUT "YOUR NAME?";H1$:INPUT "PLAYER 2's NAME?";H2$
1200 GOSUB1080:PRINT H1$;:BEEP:INPUT " DO YOU WANT TO GO FIRST? Y/N";R$
1210 GOSUB 1080:R$=LEFT$(R$,1):IF R$="Y" THEN 1270
1220 IF R$<>"N" THEN 1200
1230 GOTO 1330
1240 BEEP:PRINT "NO LEGAL MOVES."
1250 BEEP:PRINT "GAME IS A DRAW."
1260 GOTO 830
1270 SCREEN 2,2:GOSUB 1060:BEEP:PRINT H1$;:PRINT " YOUR MOVE?";:R$=INKEY$
1280 IF INKEY$="" THEN 1280
1290 R$=INKEY$
1291 PRINT R$
1300 M=INT(VAL(R$)):IF M>6ORM<1 THEN GOSUB 390
1310 FOR J=1 TO Q:T(J)=B(J):NEXT
1312 GOSUB 400:IF M<0 THEN GOSUB 390
1313 FOR J=1 TO Q:B(J)=T(J):NEXT
1320 MN=MN+1:GOSUB 910
1321 IF B(P)<24 THEN 1330
1322 GOSUB 1070:BEEP:PRINT "YOU WIN!";:GOTO 820
1330 SCREEN 2,2:GOSUB 1060:BEEP:PRINT H2$;:PRINT " YOUR MOVE?";:GOSUB 1400
1340 M=INT(VAL(Z$)):IF M>12ORM<7 THEN GOSUB 390
1350 FOR J=1 TO Q:T(J)=B(J):NEXT
1360 GOSUB 400:IF M<0 THEN GOSUB 390
1370 FOR J=1 TO Q:B(J)=T(J):NEXT
1380 MN=MN+1:GOSUB 910
1381 IF B(Q)<24 THEN 1270
1390 GOTO 1322
1400 GOSUB 1410:PRINT K$;:K1$=K$:GOSUB 1430:GOSUB 1410:PRINT K$:Z=INT(VAL(K1$+K$)):Z$=MID$(STR$(Z),2):RETURN
1410 K$=INKEY$:IF K$="" THEN 1410
1420 RETURN
1430 FOR I=1 TO 30:NEXT :RETURN

```

STOP PRESS

JUST ARRIVED

**SPEECH PROCESSOR UNIT FOR
THE SEGA 3000 \$235**

Includes cable, disk, software and instructions.

It's simply brilliant. Amaze your friends.

SEGA SOFTWARE SUPPORT
P.O. BOX 10-207
AUCKLAND, 4, N.Z.

How to get HuCal to work with a Printer

Some people, those lucky to have HuCal, an SF-7000 and a Centronics printer have been having problems with the machine not printing out data properly. So here is a little program, that when run will alter your disc so that it will work with a printer.

To use it, just enter the program, run it, place your offending disc in the drive and follow the instructions.

I'd like to thank Mr Harold Webber for sending the program to me.

If anyone knows how to get SegaBase working with a printer then please send a copy of the program to us.

```
10 CLS:E=0 :CONSOLE ,,,,2
20 PRINT "          HuCAL PATCH PROGRAM"
25 PRINT "          by trevor Yann"
27 PRINT
30 PRINT "This program will alter a disk"
40 PRINT "containing HuCAL so that"
50 PRINT "any printer can be connected to the centronics port"
80 PRINT"INSERT HuCAL SYSTEM DISK AND"
90 PRINT "PRESS <CR>."
100 T$="*":C=10
110 A$=INKEY$
120 IF A$=CHR$(13) THEN 170
130 IF E=1 THEN IF A$<>" " THEN PRINT CHR$(8):END
140 C=C+1:IF C<5 THEN 110
150 PRINT T$;:C=0:IF T$="*" THEN T$=CHR$(8):GOTO 110
160 T$="*":GOTO 110
170 PRINT CHR$(8);:BEEP
180 TR=0:SC=1
190 DSKI$ TR,SC;A$(0),0,128;A$(1),128,128
200 IF LEFT$(A$(0),18)<>"SYS: HuCAL SC-3000" THEN BEEP2:PRINT :PRINT "THIS IS NO
T A HuCAL SYSTEM DISK.":PRINT:GOTO 80
210 PRINT :PRINT "Updating disk"
220 TR=0:SC=12
230 DSKI$ TR,SC;A$(0),0,128;A$(1),128,128
240 BT=&HA3
250 NW$=CHR$(&H0D)+CHR$(&H28)+CHR$(&H05)+CHR$(&HFE)+CHR$(&H0A)+CHR$(&HC2)+CHR$(&
HBB)+CHR$(&H0A)+CHR$(&H3E)+CHR$(&H0A)
260 NW$=NW$+CHR$(&HCD)+CHR$(&HCD)+CHR$(&H0A)+CHR$(&H3E)+CHR$(&H0D)+CHR$(&HCD)+CH
R$(&HCD)+CHR$(&H0A)+CHR$(&HC3)+CHR$(&HDB)+CHR$(&H0A)
270 LN=LEN(NW$)
280 A$(INT(BT/128))=LEFT$(A$(INT(BT/128)),BTMOD128)+NW$+RIGHT$(A$(INT(BT/128)),1
28-BTMOD128-LN)
290 DSKD$ TR,SC;A$(0),0,128;A$(1),128,128
300 BEEP:E=1
310 PRINT "DISK UPDATED":PRINT
320 PRINT "PRESS <cr> TO UPDATE ANOTHER DISK, ",
330 PRINT"OR ANY OTHER KEY TO END PROGRAM"
340 GOTO 100
```

DISK DRIVE UNITS SF-7000

A few now available. \$550 + \$6.50 postage and handling.

SEGA SOFTWARE SUPPORT, P.O. BOX 10-207, AUCKLAND, 4, N.Z.

The Fish Tank by the Turner Brothers

I'm not going to tell you what this program does, just run it and find out for yourself!
Its a nice little graphics demo.

```
10 SCREEN 2,2:CLS
20 CIRCLE(125,95),16,1,.5,.6,.41
21 CIRCLE( 95,95),16,1,.8,.82,.95
22 CIRCLE( 95,95),16,1,.8,.05,.18
23 CIRCLE( 90,95),16,1,1 ,.85,.15
24 CIRCLE(133,93), 2,1,1 ,1 ,1,B
25 CIRCLE(132,95),5,1,2,.4 ,.6
26 CIRCLE(122,95),10,1,1.5,.75,.9
27 CIRCLE(117,95), 8,1,2,.81,.92
28 CIRCLE(122,95),12,1,1,.12,.25
29 CIRCLE(112,100),12,1,1,.05,.1
30 CIRCLE(145,90 ),2,4,1,1,1
31 CIRCLE(145,80 ),3,4,1,1,1
32 CIRCLE(145,65 ),5,4,1,1,1
40 LINE(1,1)-(254,190),1,B
41 LINE(5,5)-(249,185),1,B
42 LINE(15,15)-(239,175),1,B
43 PAINT(6,6),15
44 LINE(1,1)-(254,190),1,B
45 Y=168
46 FOR I=16TO238
47 LET A=INT(RND(1)*3)
48 IF A=1THENY=Y-1:IFY<166THENY=Y+2
49 IF A=2THENY=Y+1:IFY>=174THENY=Y-2
50 LINE(I,175)-(I,Y),10
51 NEXT I
59 Y=168
60 FOR I=16TO238STEP 2:IFI<38 THEN A=2:GOTO 62
61 LET A=INT(RND(1)*3)
62 IF A=1THENY=Y-3:IFY<116THENY=Y+3
63 IF A=2THENY=Y+3:IFY>=169THENY=Y-3
64 LINE(I,170)-(I,Y),2
65 NEXT I
66 LINE (25,14)-(25,155),15
70 LINE (30,14)-(30,155),15
80 CIRCLE (27,160),7 ,4,1,1,1,BF
90 FOR A=1TO 70
100 LETX=INT(RND(1)*25)+16:LET Y=INT(RND(1)*139)+16
110 PSET (X,Y),15
120 NEXT A
130 LINE (230,18)-(230,90),15
140 LINE (225,18)-(225,90),15
150 LINE (225,90)-(227,95 ),15
160 LINE (230,90)-(228,95 ),15
170 CIRCLE (227 ,97),2,9,1,1,1,BF
180 LINE (225,18)-(227,16),15
190 LINE (230,18)-(228,16),15
200 LINE (16 ,18)-(236,18),4
205 PATTERNS#0,"4040404040404040"
210 MAGO
211 S=0
215 FOR I=85TO55STEP-5:SPRITES,(227,I),0,6:S=S+1:NEXT I
231 COLOR,7,(0,0)-(255,191)
240 GOTO 240
```

A plea. If you write to us, make sure you include your name and address... please... please.

Code Breaker by Rod Cuckow

This is a simulation of the famous Mastermind game. Luckily, the instructions are in the game.

```
1 REM          CODE BREAKER
.
10 CLS:COLOR 1,14
20 SCREEN 2,1:CLS:COLOR 4,14,,14
30 CURSOR 100,40:PRINT CHR$(17);"SEGA"
40 CURSOR 55,55:PRINT "CODE BREAKER"
50 CURSOR 120,80:PRINT CHR$(16);"by"
60 PRINT :PRINT :COLOR 6:PRINT CHR$(17);          "  d"
70 PRINT "    d ROD CUCKOW "
80 PRINT "    d"
90 SCREEN 2,2
100 GOSUB 1590
110 FOR DE=1 TO 100:NEXT
120 SOUND 0
130 SCREEN 1,1
140 CURSOR 3,10:PRINT "DO YOU NEED INSTRUCTIONS? (Y/N)"
150 K$=INKEY$:IF K$="N" THEN COLOR 1:GOTO 310
160 IF K$="Y" THEN COLOR 1:GOTO 180
170 C=INT(RND(1)*13):COLOR C:GOTO 140
180 CLS:CURSOR 3,2:PRINT "The object of the game is to"
190 PRINT TAB(5);"try and work out the code"
200 PRINT TAB(5);"of four colours and blanks"
210 PRINT TAB(2);"hidden behind the question marks"
220 PRINT TAB(2);"There are a total of six colours"
230 PRINT TAB(13);"plus blanks."
240 PRINT TAB(5);"The results of each try are"
250 PRINT TAB(7);"shown by black and white"
260 PRINT TAB(5);"pegs: BLACK = right colour"
270 PRINT TAB(5);"wrong place, WHITE = right"
280 PRINT TAB(9);"colour right place"
290 PRINT :PRINT TAB(9);"YOU HAVE TEN TRIES"
300 FOR DE=1 TO 1000:NEXT
310 CURSOR 3,16:PRINT "PRESS ANY KEY TO START"
350 K$=INKEY$:IF K$<>" " THEN BEEP:GOTO 400
360 IF K$="2" THEN BEEP:GOTO 500
370 IF K$="3" THEN BEEP:GOTO 530
380 IF K$="4" THEN BEEP:GOTO 550
390 GOTO 350
400 GOSUB 740
410 C(1)=R
420 GOSUB 740
430 C(2)=R:IF C(2)=C(1) THEN 420
440 GOSUB 740
450 C(3)=R:IF C(3)=C(1)ORC(3)=C(2) THEN 440
460 GOSUB 740
470 C(4)=R:IF C(4)=C(3)ORC(4)=C(2)ORC(4)=C(1) THEN 460
480 FOR I=1 TO 4:E(I)=C(I):NEXT
490 GOTO 750
500 FOR I=1 TO 4:GOSUB 740
510 C(I)=R:NEXT
520 GOTO 480
530 FOR I=1 TO 4:R=INT(RND(1)*7+1):C(I)=R:NEXT
540 GOTO 480
550 CLS:FOR I=1 TO 4
560 CURSOR 3,10:PRINT "ENTER FOUR COLORS":PRINT
570 PRINT TAB(9);"(R) for RED"
580 PRINT TAB(9);"(G) for GREEN"
590 PRINT TAB(9);"(B) for BLUE"
600 PRINT TAB(9);"(Y) for YELLOW"
605 PRINT TAB(9);"(V) for VIOLET"
610 PRINT TAB(9);"(W) for WHITE"
620 PRINT TAB(9);"(S) for BLANK"
630 K$=INKEY$:IF K$="R" THEN C(I)=1
640 IF K$="G" THEN C(I)=2
650 IF K$="B" THEN C(I)=3
```

```


660 IF K$="Y" THEN C(I)=4
670 IF K$="V" THEN C(I)=5
680 IF K$="W" THEN C(I)=6
690 IF K$="S" THEN C(I)=7
700 IF K$="R"OR K$="G"OR K$="B"OR K$="Y"OR K$="V"OR K$="W"OR K$=" " THEN 720
710 GOTO 630
720 BEEP:PRINT :PRINT TAB(14);I;" ";K$;:FOR J=1 TO 30:NEXT :NEXT
730 PRINT TAB(18);"thanks":GOTO 480
740 R=INT(RND(1)*7):RETURN
750 SCREEN 2,2:CLS:PRINT CHR$(16)
760 COLOR 4,15,(176,31)-(253,191),14
765 LINE (33,31)-(176,31),15
770 LINE (33,10)-(253,191),15,B
780 LINE (134,10)-(134,191),15
790 LINE (175,10)-(175,191),15
800 REM :PAINT(140,15),15
810 COLOR4,14
820 CURSOR 46,19:PRINT "? ? ? ?"
830 CURSOR 188,14:PRINT "Use these"
840 PRINT TAB(32);"keys"
850 PRINT
860 PRINT TAB(29);"R = red"
870 PRINT TAB(29);"G = green"
880 PRINT TAB(29);"B = blue"
890 PRINT TAB(29);"Y = yellow"
900 PRINT TAB(29);"V = violet"
910 PRINT TAB(29);"W = white"
920 PRINT TAB(29);"S = blank"
930 PRINT
940 PRINT TAB(29);"CURSOR KEYS"
950 PRINT :PRINT TAB(29);"LEFT= back"
960 PRINT TAB(29);"DOWN=line ok"
970 SCREEN 2,2
980 Y=45:A=1
990 X=46:FOR I=1 TO 4
1000 IF X<46 THEN X=46:I=1
1010 FOR J=1 TO 20:NEXT
1020 CURSOR X,19:COLOR 1:PRINT "?"
1030 K$=INKEY$:IF K$="" THEN CURSOR X,19:COLOR 15:PRINT "?":GOTO 1020
1040 CURSOR X,11:PRINT CHR$(8)
1050 IF K$="R" THEN P(I)=1:Q=8
1060 IF K$="G" THEN P(I)=2:Q=12
1070 IF K$="B" THEN P(I)=3:Q=4
1080 IF K$="Y" THEN P(I)=4:Q=10
1090 IF K$="V" THEN P(I)=5:Q=13
1100 IF K$="W" THEN P(I)=6:Q=15
1110 IF K$="S" THEN P(I)=7:Q=0
1120 IF K$=CHR$(29) THEN X=X-24:I=I-1:GOTO 1000
1130 IF K$=CHR$(31) THEN 1210
1140 IF K$="R"OR K$="G"OR K$="B"OR K$="Y"OR K$="V"OR K$="W"OR K$="S" THEN 1160
1150 GOTO 1020
1160 IF Q=0 THEN CIRCLE(X,Y),5,1
1165 IF Q<>0 THEN CIRCLE(X,Y),5,Q,,,BF
1170 IF I=4 THEN FOR T=1 TO 15:LINE (188,134)-(240,134):NEXT
1180 X=X+24:IF X>118 THEN X=118
1190 IF I=4 THEN 1020
1200 NEXT
1210 B=0:D=0:W=140
1220 FOR I=1 TO 4
1230 IF P(I)=C(I) THEN D=D+1:P(I)=8:C(I)=9
1240 NEXT I
1250 FOR I=1 TO 4
1260 FOR J=1 TO 4
1270 IF P(I)=C(J) THEN B=B+1:P(I)=10:C(J)=11
1280 NEXT J
1290 NEXT I
1300 IF B=0 THEN 1320
1310 FOR I=1 TO B:CURSOR W,Y:COLOR1,14:PRINT "f":W=W+8:NEXT I
1320 IF D=0 THEN 1360
1330 FOR I=1 TO D:CURSOR W,Y:COLOR15,14:PRINT "T":W=W+8:NEXT I
1340 IF D=4 THEN BEEP:CURSOR 181,139:COLOR 12,14:PRINT "YOU TOOK";A:CURSOR 196,1

```

```

48:PRINT "TRIES":GOSUB 1400:GOSUB 1690
1350 IF D=4 THEN FOR F=15 TO 1 STEP -.5:SOUND 1,300-12*F,F:NEXT F:SOUND 0:CURSOR
 40,2:PRINT CHR$(17);"CORRECT";CHR$(16):GOTO 1400
1360 FOR I=1 TO 4:C(I)=E(I):NEXT I
1370 IF A=10 THEN 1390
1380 Y=Y+15:A=A+1:BEEP:GOTO 990
1390 CURSOR 44,2:COLOR12,14:PRINT "SORRY YOU LOSE":FOR I=1 TO 50:SOUND 1,110,12:
NEXT :SOUND 0
1400 X=46
1410 FOR I=1 TO 4
1420 IF E(I)=1 THEN Q=8
1430 IF E(I)=2 THEN Q=12
1440 IF E(I)=3 THEN Q=4
1450 IF E(I)=4 THEN Q=10
1460 IF E(I)=5 THEN Q=13
1470 IF E(I)=6 THEN Q=15
1480 IF E(I)=7 THEN Q=0
1490 CIRCLE(X,20),8,Q,1.1,0,1,BF
1500 X=X+24:NEXT
1510 CURSOR 184,160:COLOR6,14:PRINT "DO YOU WANT"
1520 PRINT TAB(30);"TO PLAY"
1530 PRINT TAB(30);"AGAIN Y/N"
1540 K$=INKEY$
1550 IF K$="N" THEN 1580
1560 IF K$="Y" THEN ERASE:SCREEN 2,2:CLS:GOTO 400
1570 GOTO 1540
1580 CLS:CURSOR 58,40:PRINT CHR$(17);"SOME PEOPLE":CURSOR 35,60:PRINT "CAN'T TAK
E IT !!!":FOR I=1 TO 1000:NEXT :SCREEN 1,1:CLS:END
1590 FOR K=1 TO 2:RESTORE
1600 FOR DE=1 TO 100:NEXT
1610 FOR J=1 TO 28:READ T
1620 IF T=0 THEN 1640
1630 FOR I=1 TO T:SOUND 1,1000,12:NEXT
1640 SOUND 0
1650 FOR DE=1 TO 20:NEXT
1660 NEXT J,K
1670 DATA 2,2,2,0,2,0,6,6,2,0,2,6,0,0,0,0,6,6,0,2,2,0,6,2,0,6,2,2
1680 RETURN
1690 SCREEN 1:CLS:COLOR1,15:FOR I=1 TO 6:SCREEN 1,1:SCREEN 2,2:NEXT :RETURN

```



The Case by Rod Cuckow

In this game, you play the part of a secret agent trying to find some important documents, but like any game of this type it is exceptionally complex and great fun. A must for budding Bogarts!

Instructions are in the program.

'the case'



```
1 REM THE CASE
   by
   ROD CUCKOW.
2 REM
3 SCREEN 2,2:CLS:COLOR 1,13:CURSOR50,50:PRINT CHR$(17);"YOU NEED PAPER":CURSOR70
,70:PRINT "AND PENCIL":CURSOR30,90:PRINT "TO PLAY THIS GAME":CURSOR 60,180:PRINT
CHR$(16);"PRESS ANY KEY TO START"
4 IF INKEY$="" THEN 4
5 SCREEN 2,2:CLS:GOSUB 150
6 ERASE
7 GOSUB 117
8 DIMBT$(100),SH$(100)
9 P=1:Y=0:X=1:EP=10:SU=0:OP=0
10 U=0:NM=0:Z=16:BT=0:SH=0:FT=0:S$="":B$=""
11 PRINT CHR$(19)
12 CLS
13 SCREEN 1,1:CLS:COLOR15,13:CURSOR 15,0:PRINT"THE CASE"
14 CURSOR 14,1:PRINT"=====":NM=NM+1
15 DT=0:CURSOR 0,2:PRINT "You are at the":CURSOR 24,2:PRINT "Time is";STR$(H)+ST
R$(M)
16 IF H=9 THEN CURSOR 33,3:PRINT "' ' "
17 IF H>9 THEN CURSOR 34,3:PRINT "' ' "
18 IF H>=Z AND M>15 AND F(6)=0 THEN CURSOR 2,21:PRINT "TO LATE YOUR CONTACT HAS
GONE HOME":GOSUB 236:FOR DE=1 TO 100:NEXT:CURSOR 2,21:PRINT "
"
19 A=19-(LEN(B$)/2):CURSOR A,19:PRINT B$:B$=""
20 I=7-(LEN(R$(P))/2)
21 IF BT$(BT)=R$(P) THEN BT=BT-1
22 IF BT$(BT)=R$(P) THEN BT=BT-1
23 BT=BT+1:CURSOR I,4:PRINT R$(P):BT$(BT)=R$(P)
24 IF EP=P THEN CURSOR 0,21:PRINT "ENEMY AGENT IS HERE":NE=NE+1
25 IF EP<>P THEN NE=0
26 IF P=MP AND F(1)=0 THEN CURSOR 24,4:PRINT "MESSAGE HERE"
27 F(3)=0:T1=H+M/100
28 IF F(4)=1 AND R$(P)=S$ AND U<=T1 AND U+.15>T1 THEN CURSOR 23,4:PRINT "CONTACT
IS HERE":F(3)=1
```



```

29 IF P=1 AND H<FH AND F(7)=1 THEN GOTO 103
30 CURSOR 14,6:PRINT "WHAT NEXT":CURSOR 4,8:PRINT"(hold down X for status report
)"
31 CURSOR 17,10:PRINT " "
32 IF INKEY$="" THEN CURSOR 17,10:PRINT "? e":GOTO 32
33 IF INKEY$="X" THEN GOTO 239
34 CURSOR 17,10:INPUT I$
35 IF I$="END" THEN GOTO 222
36 IF I$="FOLLOW" AND SU>5 THEN GOTO 46
37 IF I$="ESP" THEN CURSOR 17,4:PRINT LEFT$(R$(EP),1);"?";RIGHT$(R$(EP),1):CURSO
R 17,10:PRINT "? "":GOTO 31
38 IF I$<>"HIDE" THEN 41
39 P=FNA(20):IF P=EP THEN 39
40 SU=0:GOTO 13
41 V=0:FOR I=1 TO 10
42 IF I$=V$(I) THEN V=I
43 NEXT
44 IF V=0 THEN V=11
45 IF NE>3 AND FNA(10)>3 AND V<>1 AND EP=P THEN CURSOR 22,21:PRINT "AND SEES YO
U": SU=SU+1:GOTO 14
46 IF SU>2 THEN CLS:CURSOR 10,3:PRINT "YOU ARE CAPTURED":SU=0:GOSUB 207:GOTO 13
47 ON V GOSUB 52,59,66,73,76,82,89,92,98,135,175
48 M=M+DT:IF M>59 THEN M=M-60:H=H+1
49 IF F(2)=1 AND H>=CH THEN F(4)=1
50 IF H=FH THEN GOTO 102
51 GOTO 13
52 CURSOR 15,12:PRINT"WHERE TO"
53 CURSOR 17,14:INPUT N$
54 NP=0:FOR I=1 TO 20
55 IF N$=R$(I) THEN NP=I
56 NEXT:IF NP=0 THEN GOSUB 175:GOTO 13
57 GOSUB 111
58 P=NP:RETURN
59 DT=5
60 CURSOR 15,12:PRINT "SAY WHAT":CURSOR 17,14:INPUT Q$
61 IF EP=P THEN B$="YOU ATTRACTED THE ENEMY AGENT":IF FNA(7)>4 THEN SU=3:RETURN
62 IF F(3)=0 THEN B$="NOBODY HEARS YOU":RETURN
63 IF Q$<>P$ THEN B$="CONTACT IGNORES YOU":RETURN
64 IF F(6)=1 THEN B$="YOU MADE CONTACT - HE TAKES THE CASE"
65 F(7)=1:RETURN
66 DT=5
67 CURSOR 6,12:PRINT "WHAT DO YOU WANT TO EXAMINE":CURSOR 17,14:INPUT Q$
68 IF Q$="CASE"AND F(6)=0 THEN B$="WHAT CASE?":RETURN
69 IF Q$="KEY"AND F(5)=0 THEN B$="WHAT KEY?":RETURN
70 IF Q$="CASE" THEN B$="TOP SECRET":RETURN
71 IF Q$="KEY" THEN B$="A NUMBER -"+STR$(KN):RETURN
72 B$="NOTHING SPECIAL":RETURN
73 IF P<>MP OR F(1)=1 THEN B$="NOTHING TO READ":RETURN
74 B$="A WORD - '"+P$+"'"
75 F(1)=1:RETURN
76 DT=5
77 IF P<>16 THEN B$="NOTHING TO OPEN":RETURN
78 IF F(5)=0 THEN B$="YOU DO NOT HAVE A KEY":RETURN
79 CURSOR 10,12:PRINT "WHAT LOCKER NUMBER":CURSOR 17,14:INPUT LN
80 IF KN<>LN THEN B$="THE KEY DOES NOT FIT":RETURN
81 B$="THE LOCKER IS OPEN - YOU HAVE THE CASE":F(6)=1:RETURN
82 DT=5
83 IF EP<>P THEN B$="FOLLOW WHO?":RETURN
84 NP=FNA(20):GOSUB 111:P=NP
85 IF FNA(10)>8 THEN P=KP
86 IF FNA(10)> 8 AND EP<>10 THEN B$="HE GIVES YOU THE SLIP":RETURN
87 EP=P
88 B$="":RETURN
89 CURSOR 11,12:PRINT "HOW MANY MINUTES":CURSOR 17,14:INPUT DT:IF DT<0 THEN CURS
OR 24,15:PRINT "CHEAT":BEEP:CURSOR 17,14:PRINT " "":GOTO 89
90 IF DT>60 THEN CURSOR 0,15:PRINT "NO MORE THAN - 60 - AT A TIME":BEEP:CURSOR 1
7,14:PRINT " "":GOTO 89
91 RETURN
92 IF H=CH AND M>00 THEN CURSOR 2,17:PRINT "TO LATE YOUR CONTACT HAS GONE HOME":
GOTO 222
93 CURSOR 7,12:PRINT "WHERE DO YOU WANT TO MEET ":CURSOR 17,14:INPUT S$
94 CURSOR 0,15:PRINT :INPUT "WHAT TIME (HH.MM) "":U:Z=U

```

```

95 IF CH>=Z OR Z>FH THEN CURSOR 6,17:PRINT "DO YOU HAVE A TIME MACHINE ?":BEEP:G
OTO 94
96 IF P=CP AND T1<U AND H<CH THEN F(2)=1
97 DT=5:RETURN
98 B$="NOTHING HERE":DT=10:SH=SH+1:SH$(SH)=R$(P)
99 IF F(5)=1 AND P=KP THEN B$="NOTHING HERE NOW"
100 IF P=KP AND F(5)=0 THEN B$="YOU FOUND A KEY":F(5)=1
101 RETURN
102 CLS:PRINT :PRINT "TOO LATE - YOU MISSED THE LAST PLANE":FORI=1TO200:NEXT:SOU
ND4,3,15:FORI=1TO9:SOUND3,3500:NEXT:FORI=15TOOSTEP-.2:SOUND4,,I:NEXT:PRINT:PRINT
" AND THE ENEMY AGENT SHOT YOU":GOTO 222
103 CURSOR 7,12:PRINT "YOUR MISSION WAS A SUCCESS"
104 TL=(FH-H)*60-M
105 R=INT((20/NM+TL/120)*50)
106 CURSOR13,15:PRINT "YOUR SPY RATING"
107 GOSUB 225
108 CURSOR13,17:PRINT " IS"
109 CURSOR13,19:PRINT " ";R
110 GOTO 222
111 NY=INT((NP-1)/5)
112 NX=NP-5*NY
113 DX=ABS(X-NX):DY=ABS(Y-NY)
114 X=NX:Y=NY
115 D=SQR(DX^2+DY^2)
116 DT=INT(5*D):RETURN
117 DIM R$(20),V$(10)
118 FOR I=1 TO 20:READ R$(I):NEXT
119 FOR I=1 TO 10:READ V$(I):NEXT
120 DATA AIRPORT,BUS STOP,BRIDGE,RIVER,CHURCH
121 DATA PARK,CAFE,BANK,CINEMA,HOTEL
122 DATA MARINA,NIGHT CLUB,POST OFFICE,POLICE STATION,FAIRGROUNDS
123 DATA STATION,TOWN HALL,LIBRARY,COLLEGE,GARAGE
124 DATA MOVE,SAY,EXAMINE,READ,OPEN,FOLLOW,WAIT,LEAVE,SEARCH,HELP
125 DATA CASE,CHASE,SEGA,KEYBOARD,POP GROUP,GOOD GAME
126 DEF FNA(X)=INT(RND(1)*X+1)
127 H=FNA(2)+8:M=0
128 FH=FNA(2)+14
129 CH=FNA(2)+H
130 MP=FNA(18):KP=FNA(18)
131 EP=FNA(18):CP=FNA(18)
132 FOR I=1 TO FNA(6)
133 READ P$:NEXT
134 KN=FNA(900)+99
135 SCREEN 1,1:CLS
136 PRINT TAB(13);"THE - CASE"
137 PRINT TAB(13);"=====":PRINT
138 PRINT "COLLECT MESSAGE FROM":PRINT
139 PRINT "THE ";R$(MP):PRINT
140 PRINT "CONTACT WILL COLLECT":PRINT
141 PRINT "FROM THE ";R$(CP):PRINT
142 PRINT "AT ";CH;".00"
143 PRINT:PRINT:PRINT "LAST FLIGHT LEAVES":PRINT
144 PRINT "AT ";FH;".00":PRINT
145 IF FT=0 THEN 147
146 PRINT "every time you use HELP you waste time and increase the chance of ge
tting caught"
147 PRINT :PRINT "press any key to continue":FT=1
148 IF INKEY$="" THEN 148
149 DT=INT(RND(1)+10*NM):SU=SU+.5:RETURN
150 CURSOR 80,90:COLOR 1:PRINT CHR$(17);"THE":CURSOR 120,120:PRINT "CASE";CHR$(1
6)
151 SOUND 5,3,15:FOR I=1100 TO 110 STEP-70:SOUND 3,1,10:NEXT :SOUND 0
152 SOUND 4,0,0:FOR I=15 TO 0 STEP-1:SOUND 1,900,I:SOUND 4,,I:NEXT
153 SOUND 4,0,15:FOR I=1 TO 15 STEP .4:SOUND 4,,I:NEXT :FOR I=15 TO 0 STEP -1:SO
UND4,,I:NEXT
154 SOUND1,,15:SOUND2,,15:SOUND3,,15:FORI=2000TO1175STEP-75:SOUND1,I:SOUND2,2220
-I:SOUND3,(IXOR1100)+100:NEXT:FORI=15TOOSTEP-1:SOUND1,,I:SOUND2,,I:SOUND3,,I:NEX
T
155 SOUND3,,0:FOR A=10000 TO 500 STEP -330:SOUND3,A:SOUND4,3,15:SOUND5,3,15:NEXT
:FOR I=14 TO 0 STEP -1:SOUND5,3,I:NEXT
156 FOR I=510 TO 4000 STEP 200:SOUND4,3,15:SOUND3,1,9:NEXT :FOR I=15 TO 0 STEP -
.5:SOUND4,,I:NEXT :SOUND0

```

```

157 SOUND4,3,15:FOR I=1 TO 20:SOUND3,RND(8)*3000+500:NEXT :FOR I=15 TO 0 STEP -.3
:SOUND4,,I:NEXT
158 SOUND 5,3,15:FOR I=1 TO 20:SOUND3,RND(8)*3000+500:NEXT :FOR I=15 TO 0 STEP -
.3:SOUND5,,I:NEXT
159 FOR DE=1 TO 120:NEXT
160 SOUND4,3,15:FOR I=1 TO 9:SOUND3,3500:NEXT :FOR I=15 TO 0 STEP -.2:SOUND4,,I:N
EXT
161 FOR DE=1 TO 75:NEXT
162 SCREEN 1,1:CLS
163 PRINT :PRINT :PRINT " -----":PRINT "Your Mis
sion is a complicated one, so read these instructions carefully.":PRINT :PRINT "
You must collect a case from a locker at the station, hand it over to your cont
act and get";
164 PRINT " back to the airport before the last plane takes off (the com
puter will tell you what time)":PRINT :PRINT :PRINT "press a key to continue"
165 IF INKEY$="" THEN 165
166 PRINT :PRINT:PRINT
167 PRINT "The computer will tell you where your contact will be at what time. Y
ou must leave a message at that place, before he gets there, telling him where an
d when you will meet him to hand over the case."
168 PRINT " You must find out the password before you meet him, and make sure you
are not more than 15 minutes late.":PRINT " Before you can get the case, you
must find the key to the locker and also its number. Unfortunately the key is"
169 PRINT "in the hands of enemy agents, whose HQ is at the Hotel. You must fi
nd an enemy spy and follow him, hoping he will be careless enough to drop the
key (and not see you). If he does see you. You will have to fight. If you l
oose him";
170 PRINT " try 'esp'. ----"
171 PRINT :PRINT :PRINT "press a key to continue"
172 IF INKEY$="" THEN 172
173 GOSUB 175
174 GOTO 6
175 SCREEN 2,2:CLS
176 CURSOR 70,10:PRINT CHR$(17);"THE CASE"
177 CURSOR 40,30:COLOR14:PRINT CHR$(16);"here is a list of places to go"
178 CURSOR 12,50:COLOR 1:PRINT CHR$(17);"BANK POLICE STATION"
179 CURSOR 12,60:PRINT "PARK POST OFFICE"
180 CURSOR 12,70:PRINT "CAFE FAIRGROUNDS"
181 CURSOR 12,80:PRINT "RIVER NIGHT CLUB"
182 CURSOR 12,90:PRINT "HOTEL TOWN HALL"
183 CURSOR 12,100:PRINT "BRIDGE BUS STOP"
184 CURSOR 12,110:PRINT "CHURCH AIRPORT"
185 CURSOR 12,120:PRINT "GARAGE STATION"
186 CURSOR 12,130:PRINT "CINEMA LIBRARY"
187 CURSOR 12,140:PRINT "MARINA COLLEGE"
188 CURSOR 50,150:COLOR14:PRINT CHR$(16);"press any key to continue ":PRINT :PRI
NT " I suggest you write down the places"
189 IF INKEY$="" THEN 189
190 SCREEN 2,2:CLS
191 CURSOR 70,0:COLOR15:PRINT CHR$(17);Z$
192 CURSOR 40,20:COLOR14:PRINT CHR$(16);"here is a list of words to use"
193 CURSOR 40,40:COLOR 1:PRINT CHR$(17);"SAY EXAMINE"
194 CURSOR 40,50:PRINT "MOVE FOLLOW"
195 CURSOR 40,60:PRINT "READ SEARCH"
196 CURSOR 40,70:PRINT "OPEN LEAVE"
197 CURSOR 40,80:PRINT "WAIT HELP"
198 CURSOR 40,90:PRINT "HIDE END"
199 CURSOR 50,110:COLOR14:PRINT CHR$(16);"here is a list of passwords"
200 CURSOR 40,130:COLOR1:PRINT CHR$(17);"CASE KEYBOARD
201 CURSOR 40,140:PRINT "SEGA GOOD GAME"
202 CURSOR 40,150:PRINT "CHASE POP GROUP"
203 CURSOR 50,170:COLOR14:PRINT CHR$(16);"write them down then"
204 CURSOR 50,180:PRINT "press any key to start"
205 IF INKEY$="" THEN 205
206 SCREEN 2,2:CLS:RETURN
207 PRINT :PRINT " YOU FIGHT WITH THE AGENT":PRINT
208 SOUND 5,3,15:FOR I=1100 TO 110 STEP-70:SOUND 3,I,10:NEXT :SOUND 0
209 SOUND 4,0,0:FOR I=15 TO 0 STEP-1:SOUND 1,900,I:SOUND 4,,I:NEXT
210 SOUND 4,0,15:FOR I=1 TO 15 STEP .4:SOUND 4,,I:NEXT :FOR I=15 TO 0 STEP -1:SO
UND4,,I:NEXT
211 SOUND1,,15:SOUND2,,15:SOUND3,,15:FOR I=200 TO 1175 STEP-75:SOUND1,I:SOUND2,2220
-I:SOUND3,(IXOR1100)+100:NEXT:FOR I=15 TO 0 STEP-1:SOUND1,,I:SOUND2,,I:SOUND3,,I:NEXT

```

```


212 SOUND3,,0:FOR A=10000 TO 500 STEP -330:SOUND3,A:SOUND4,3,15:SOUND5,3,15:NEXT
:FOR I=14 TO 0 STEP -1:SOUND5,3,I:NEXT
213 FOR I=510 TO 4000 STEP 200:SOUND4,3,15:SOUND3,I,9:NEXT :FOR I=15 TO 0 STEP -
.5:SOUND4,,I:NEXT :SOUND0
214 SOUND4,3,15:FOR I=1 TO 20:SOUND3,RND(8)*3000+500:NEXT :FOR I=15 TO 0 STEP -.3
:SOUND4,,I:NEXT
215 SOUND 5,3,15:FOR I=1 TO 20:SOUND 3,RND(8)*3000+500:NEXT:FOR I=15 TO 0 STEP-.
3:SOUND 5,,I:NEXT
216 FOR DE=1 TO 125:NEXT
217 SOUND4,3,15:FOR I=1 TO 9:SOUND3,3500:NEXT :FOR I=15 TO 0 STEP -.2:SOUND4,,I:N
EXT
218 IF FNA(10)>7 THEN PRINT : PRINT " YOU STAGGER AWAY TO THE HOSPITAL":PRINT:
PRINT" TO RECOVER TO FIGHT ANOTHER DAY":GOTO 222
219 IF FNA(10)>4 THEN PRINT : PRINT " YOU OVERPOWER THE ENEMY AGENT": OP=1:EP
=0:IF F(5)=0 AND FNA(5)>3 THEN PRINT :PRINT " AND FIND THE KEY":F(5)=1
:FOR DE=1 TO 350:NEXT :EP=1:RETURN
220 IF OP=1 THEN FOR DE=1 TO 250:NEXT :RETURN
221 IF OP=0 THEN PRINT :PRINT " YOU ARE OVERPOWERD AND THROWN":PRINT:PRINT"
IN JAIL TO ROT":PRINT
222 PRINT :PRINT :PRINT :PRINT "Do you want to play again - Y/N ";;GOSUB 236:INP
UT A$:GOSUB 236:SU=0 :FT=0
223 IF A$="Y" THEN RESTORE :GOTO 6
224 CLS:END
225 RESTORE 235
226 FOR S=1 TO 15
227 READ L
228 SOUND1,L,15:FORN=0T035+S:NEXTN
229 SOUND2,L,15:FORN=0T035+S:NEXTN
230 SOUND3,L,15:FORN=0T035+S:NEXTN
231 NEXTS
232 FOR I=15 TO 0 STEP -.5:SOUND1,L,I:SOUND2,L,I:SOUND3,L,I:NEXT
233 SOUND 0
234 RETURN
235 DATA 262,330,392,330,262,440,494,523,494,440,392,330,392,330,262
236 FORS=300T03000STEP175:SOUND1,S,10:SOUND2,500+S,13:SOUND3,2000-(S/2),15:NEXTS
237 FOR DE=1 TO 50:NEXT :SOUND 0
238 RETURN
239 IF NM>1 THEN 244
240 IF NM=1 THEN CLS:PRINT :PRINT "ONLY PRESS X IF YOU NEED TO CHECK":PRINT:PR
INT" UP ON YOUR MOVES.":PRINT :PRINT "OTHERWISE INPUT ONE OF THESE WORDS :-":PR
INT :PRINT
241 PRINT " SAY, EXAMINE, MOVE, FOLLOW, READ,":PRINT :PRINT " SEARCH, OPEN, LEAV
E, WAIT, HELP, ":PRINT :PRINT " HIDE, END,":PRINT :PRINT "press a key to continu
e"
242 IF INKEY$="" THEN 242
243 GOTO 13
244 CLS:PRINT :PRINT ". STATUS REPORT ~~~~~~ ~~~~~~"
245 PRINT "YOU HAVE BEEN TO :-":PRINT
246 FOR I=1 TO BT
247 PRINT BT$(I)
248 FOR DE=1 TO 50:NEXT
249 NEXT I
250 PRINT :PRINT "press a key to continue"
251 IF INKEY$="" THEN 251
252 PRINT :PRINT "YOU HAVE SEARCHED :-":PRINT
253 FOR I=1 TO SH:PRINT SH$(I)
254 FOR DE=1 TO 50:NEXT
255 NEXT I
256 PRINT :PRINT "press a key to continue"
257 IF INKEY$="" THEN 257
258 PRINT :PRINT
259 IF F(1)=1 THEN PRINT "YOU HAVE THE PASSWORD :- ";P$:GOTO 261
260 PRINT "YOU DO NOT HAVE THE PASSWORD YET"
261 PRINT :PRINT
262 IF F(5)=1 THEN PRINT "YOU HAVE THE KEY NUMBER :-";KN:GOTO 264
263 PRINT "YOU DO NOT HAVE THE KEY YET"
264 PRINT :PRINT
265 IF F(6)=1 THEN PRINT "YOU HAVE THE CASE":GOTO 267
266 PRINT "YOU DO NOT HAVE THE CASE YET"
267 PRINT :PRINT "press a key to continue"
268 IF INKEY$="" THEN 268
269 GOTO 13

```

SpaceFighter by Richard Turner

In this game you are in charge of a space ship and you must destroy the red beasties on planet surface. To run use the joystick and fire to fire.

The game is actually quite addictive, and gives some insight into how you can write a good game.



```

1 SC=0: MEN=3: PRINT CHR$(16): GOSUB 6000
2 SCREEN 1,1
5 CLS: PRINT "Please Wait."
10 SCREEN 2,1: CLS: COLOR 15,1, (0,0)-(255,191)
15 COLOR,15, (210,0)-(255,191)
20 LINE (206,0)-(207,191),6,B
40 CURSOR218,10: COLOR 1: PRINT "SCORE"
50 COLOR,10, (216,8)-(240,18)
60 LINE (215,7)-(248,19),1,B
65 LINE (213,5)-(250,21),1,B
68 CURSOR218,30: PRINT SC
70 COLOR,10, (216,60)-(240,70)
71 CURSOR218,63: COLOR 1: PRINT "MEN!.."
72 LINE (215,61)-(248,71),1,B
73 LINE (213,59)-(250,73),1,B
75 CURSOR218,83: PRINT MEN
100 FOR I=0 TO 250: LET X=INT(RND(1)*210): Y=INT(RND(1)*191): PSET(X,Y),10: NEXT
101 CIRCLE(102,190),102,6,.15,1,1,BF
102 PATTERNS#5,"18183C3CFFFC3C3"
103 PATTERNS#6,"C3C3FF7EFF7E6666"
104 PATTERNS#7,"E7E7E7E7E7E7E7"
105 PATTERNS#8,"0E0F0F0E0E0E0606"
106 PATTERNS#9,"70F0F07070706060"
110 LINE(90,80)-(120,110),4,B
111 LINE(110,100)-(140,130),4,B
112 LINE(90,80)-(110,100),4
113 LINE(120,80)-(140,100),4
114 LINE(90,110)-(110,130),4
115 LINE(120,110)-(140,130),4
116 BLINE (218,29)-(250,45),1,BF
117 CURSOR218,30: PRINT SC
118 SCREEN 2,2
120 PATTERNS#0,"0000000000000F06"
121 PATTERNS#1,"1F3E7CFC1F000000"
122 PATTERNS#2,"0000000000000F060"
123 PATTERNS#3,"FB7C3E3FF8000000"
124 MAG0: X=95: Y=80
130 FOR I=1 TO 25
140 X=X+2: Y=Y+2
145 A=20 : B=161
148 LET V=0
150 SPRITE0,(X,Y),0,4
160 SPRITE2,(X+8,Y),2,4
170 SPRITE1,(X,Y+8),1,4
180 SPRITE3,(X+8,Y+8),3,4
190 NEXT I
200 ST=STICK(1)
210 IF ST=3 THEN X=X+20

```

```

211 IF ST=7 THENX=X-20
212 IF ST=1 THENY=Y-20
213 IF ST=5 THENY=Y+20
218 IFX+8<10THENX=185
219 IFX-8>185THENX=20
220 IF Y-8<1THENY=255
221 IF Y+8>155THENY=1
222 SPRITE0,(X,Y),0,4
223 SPRITE2,(X+8,Y),2,4
224 SPRITE1,(X,Y+8),1,4
225 SPRITE3,(X+8,Y+8),3,4
230 SPRITE5,(A,B),5,10
240 SPRITE6,(A,B+8),6,10
250 SPRITE7,(A,B+16),7,10
260 SPRITE8,(A-8,B+4),8,10
270 SPRITE9,(A+8,B+4),9,10
275 V=V+1:IFV=5 THEN GOTO 800
280 IFSTRIG(1)=1THENLINE(X+8,Y+8)-(X+8,B ),6:BLINE(X+8,Y+8)-(X+8,B ),1:BEEP:IF
  A+10>XANDA<X+10THEN 3000
290 G=INT(RND(1)*2):IF G=0 THEN A=A-10
300 IF G=1 THEN A=A+10
310 IFA<10THENA=185
320 IFA>185THENA=10
530 GOTO 200
800 V=0:LETQ=INT(RND(1)*3)
810 IF Q=2 THEN LINE (A,B)-(X,Y),6:BLINE (A,B)-(X,Y),0:BEEP:GOSUB 900
820 IFQ=0 THEN LINE (A,B)-(X-2,Y),6:BLINE (A,B)-(X-2,Y),6:BEEP
830 IFQ=1 THEN LINE (A,B)-(X+2,Y),6:BLINE (A,B)-(X+2,Y),6:BEEP
840 GOTO 275
900 FOR I=15TO 0STEP-1
910 CIRCLE(X,Y),15,6,1,1,1
930 SOUND5,1,I
940 NEXT I
941 BCIRCLE(X,Y),15,6,1,1,1
950 MEN=MEN-1:IFMEN=0THEN1000
960 BLINE(218,82)-(240,92),1,BF:CURSOR218,83:PRINT MEN
970 GOTO 110
1000 SCREEN 1,1:CLS:PRINT "<<<GAME OVER>>>"
1010 PRINT CHR$(16)
1020 PRINT "Another Game {Y/N}?"
1025 PRINT "Your score ";SC
1030 IF INKEY$="N" THEN END
1035 IF INKEY$="Y" THEN SC=0:MEN=3:SCREEN 2,1:GOTO 110
1040 GOTO 1030
3000 FOR I=15 TO 0 STEP -1
3010 CIRCLE(A,B),10,6,1,1,1
3015 BCIRCLE(A,B),10,6,1,1,1
3020 SOUND5,1,I
3030 NEXT I
3040 SC=SC+100
3050 BLINE (218,29)-(250,45),1,BF
3060 CURSOR218,30:PRINT SC
3070 GOTO 110
6000 SCREEN 2,2:CLS:COLOR,1,(0,0)-(255,191)
6010 CURSOR 20,10:COLOR15:PRINT "Richard Turner Presents."
6020 LINE (20,40)-(40,20),6:LINE -(40,30):LINE -(30,40):LINE -(40,50):LINE -(20,
70):LINE -(20,60):LINE -(30,50):LINE -(20,40)
6030 LINE (50,60)-(50,30),6:LINE -(70,20):LINE -(70,40):LINE -(60,45):LINE -(60,
55):LINE -(50,60)
6040 LINE (70,60)-(90,20):LINE -(110,60):LINE -(90,40):LINE -(70,60)
6050 LINE (120,60)-(100,20):LINE -(120,20):LINE -(130,30):LINE -(110,30):LINE -(
120,50):LINE -(140,50):LINE -(140,60):LINE -(120,60)
6060 LINE (150,60)-(130,20):LINE -(150,20):LINE -(160,30):LINE -(140,30):LINE -(
160,40):LINE -(165,45):LINE -(155,50):LINE -(170,50):LINE -(170,60):LINE -(150,6
0)
6070 CURSOR20,80:PRINT CHR$(17);"FIGHTER.";CHR$(16)
6080 CURSOR20,100:PRINT "(C) 1986."
6090 PAINT(23,39),6:PAINT(51,39):PAINT(91,25):PAINT(111,21):PAINT(132,21)
6100 CURSOR20,120:PRINT "Hit any key!..."
6110 IF INKEY$<>"" THEN RETURN
6120 GOTO 6110

```

DRAW!! . . . a program dissection

This is the second of our drawing programs, and is the subject of the program dissection. The program allows you to move a shape about the screen by use of the cursor keys or joystick. Pretty simple. Okay, now to the dissection.

- Line 10 This tells the computer to use the Graphics Screen, i.e. the one used for drawing lines on. If you use SCREEN 1, 1 you would tell the computer to use the text screen, you know, the one in green and black when you turn your computer on. CLS tells the computer to clear the screen, so that if anything is written on the screen, it just disappears. The instruction COLOR is a little more complex. This lets you alter the PAPER colour, the PEN colour, and the BORDER colour. The format of the instruction is PEN, PAPER, BORDER. So the command COLOR 1, 2, 3 would set the pen colour to colour 1, paper 2 and border to colour 3. When you write, you'll write in Black on a Green background and the border will change to light green. See pp 97-100
- Line 20 Make the computer print at location 80 dots across, and 90 dots down on the Graphics Screen. The message to print is "DRAW!!" The CHR\$(17) makes the computer print out the text in Double Width, so DRAW . . becomes D R A W.
- Line 30 Is a small delay. The variable L counts from 1 to 500.
- Line 40 CHR\$(16) is the opposite to CHR\$(17) so any more text will be printed in single width and not double width.
- Line 50 a REM means REMark, it has no effect on the computer program at all, it is used to keep a note for the programmer or user of the program.
- Line 60 This clears the screen.
- Line 70 This changes the pen colour to colour 1, (Black) and the border to = '(dark red).
- Line 80 This is a little complex. Firstly the computer jumps to line 310. It continues execution from there until it reaches a RETURN instruction. When the RETURN is executed, the computer jumps back to the command AFTER the GOSUB. In this case it is line 90.
- Line 90 Place the next lot of data on the screen at location X (places in) and Y (places down) the screen. The PRINT tells the computer to put a ")" on the screen at position X and Y on the screen. You can put any character between the quotes.
- Line 100 The computer now reads the keyboard, whatever key is being pressed is put into the variable A\$. So if you are pressing "Z" then the variable A\$ will hold the letter Z.
- Line 110 This is the same as saying "if the user is pressing the LEFT arrow then move the cursor left". Don't forget that the position of the cursor is determined by the variable X and Y. So if we alter one of these variables the position of the cursor will also alter. The left arrow is the same as CHR\$(29). See page 19 of the handbook.
- Line 120 As above but move right.
- Line 130 As above but move cursor up.
- Line 140 As above but move cursor down.
- Line 150 Check the status of the joystick. This data is put into the variable PL. By status I mean if the joystick has been moved or not.
- Line 160 This is the same as line 130, but the difference is that the cursor moves up if the joystick is being pushed up.
- Line 170 As above but move right.
- Line 180 As above but move down.
- Line 190 As above but move left.
- Line 200 If the joystick fire button is being pressed then clear the Screen, See page 149 (Basic Level III Manual).
- Line 210 Back to A\$. If you are pressing C then clear the screen.
- Line 220 If you are pressing "E" then end the program.

- Line 230 If X is less than 0, (eg -1 etc) then you will get an error. This is because the cursor cannot be put on the screen at position -1. So this line checks for X being less than 0, if it is X is made equal to 0. Thus preventing an error occurring.
- Line 240 If the cursor is about to fall off the right hand side of the screen, then X is brought back so that it doesn't.
- Line 250 Same as 230, but make sure the cursor doesn't go off the top of the screen.
- Line 260 Same as 240 but make sure the cursor doesn't go off the bottom of the screen.
- Line 280 Go back to line 90.
- Line 290 Print a space.
- Line 300 Print another space.
- Line 310 Print "-----" on the screen.
- Line 330 Print 205 spaces then print "Use arrows"
- Line 350 Print "Use . . ." etc on the screen.
- Line 360 A delay, T counts from 1 to 1500.
- Line 370 Set the screen of the text screen and clear it. Set the colours of the text screen to white on magenta. Note that the format for the COLOR statement is different for the two screens.
- Line 380 Clear the screen and return to the line after the GOSUB statement (line 90).

If you wish to learn more about programming then I suggest that you get hold of a copy of "Teach Yourself BASIC Games Programming" as this will help you a lot. This is available from Sega Software Support.

```

10 SCREEN 2,2:CLS:COLOR ,15,,6
20 CURSOR 80,90:PRINT CHR$(17);" DRAW !!"
30 FOR L=1 TO 500:NEXT
40 PRINT CHR$(16)
50 REM READ THIS PLEASE:YOU CAN CHANGE THE GRAPHIC CHARACTER OR PUT A LETTER INS
TEAD OF A GRAPHIC CHARACTER TRY IT!!
60 CLS
70 COLOR 1,,6
80 GOSUB 310
90 CURSORX,Y:PRINT"}"
100 A$=INKEY$
110 IF A$=CHR$(29) THEN X=X-1
120 IF A$=CHR$(28) THEN X=X+1
130 IF A$=CHR$(30) THEN Y=Y-1
140 IF A$=CHR$(31) THEN Y=Y+1
150 PL=STICK (1)
160 IF PL=1 THEN Y=Y-1
170 IF PL=3 THEN X=X+1
180 IF PL=5 THEN Y=Y+1
190 IF PL=7 THEN X=X-1
200 IF PL=8 THEN CLS
210 IF A$="C" THEN CLS
220 IF A$="E" THEN CLS:PRINT"BYE..BYE // BY ANDREW WATT//":COLOR 1,2:END
230 IFX<0 THEN X=0
240 IFX<0 THEN X=35
250 IFY>35 THEN X=35
260 IFY<0 THEN Y=0
270 IFY>22 THEN Y=22
280 GOTO 90
290 PRINT
300 PRINT
310 PRINT" INSTUCTIONS  "
320 PRINT" =====  "
330 PRINT SPC(205);"USE THE ARROW KEYS AND THE JOYSTICK TO  DRAW PICTURES AND P
RESS C TO CLEAR SCREEN"
340 PRINT
350 PRINT"USE E TO END PROGRAM. "
360 FOR T=1 TO 1500:NEXT
370 SCREEN 1,1:CLS:COLOR15,13
380 CLS:RETURN

```



Turtle

It's funny how so many people have written to me wanting a simple drawing package, just for designing small title screens, and the like. Well in this issue we have included two! This one is a drawing package, that is quite simple, but very good, and to make things even easier, the instructions are in the program.

```
1 GOSUB32
2 ONSTICK(1)GOTO6,7,8,9,10,11,12,13
3 ONSTRIG(1)GOTO14,16,18
4 IFINKEY$=" "THENGOSUB32
5 GOTO54
6 Y=Y-1:GOTO19
7 Y=Y-1:X=X+1:GOTO19
8 X=X+1:GOTO19
9 Y=Y+1:X=X+1:GOTO19
10 Y=Y+1:GOTO19
11 Y=Y+1:X=X-1:GOTO19
12 X=X-1:GOTO19
13 Y=Y-1:X=X-1:GOTO19
14 C=C+1:IFC=16THENC=1
15 GOTO22
16 HH=HH+1:IFHH=3THENHH=1
17 GOTO28
18 PRESET(X,Y):PAINT(X,Y),C
19 GOTO54
20 PSET(X,Y),C:GOTO2
21 PSET(X,Y),C:FORI=1TO10:NEXT:PRESET(X,Y):GOTO2
22 BLINE(70,0)-(130,10),,BF
23 RESTORE25
24 FORA=1TOC:READC$:NEXTA
25 DATABlack,Green,Lt Green,Dk Blue,Lt Blue,Dk Red,Cyan,Red,Lt Red,Dp Yellow,Lt
Yellow,Dk Green,Magenta,Grey,White
26 COLORC:CURSOR70,1:PRINTC$
27 GOTO19
28 IFHH=1THENHH$="Drawing":GOTO30
29 HH$="Erase"
30 BLINE(70,11)-(130,20),,BF:CURSOR70,11:COLORDC:PRINTHH$
31 GOTO19
32 SCREEN1,1:CLS:PRINT"          TURTLE GRAPHICS":PRINT:PRINT
33 PRINT" This program lets you draw you own pictures on the graphics screen u
sing a SEGA joystick. "
34 PRINT:PRINT" Move the 'Turtle' with the joystick in eight directions. The l
eft button changes the color (printed on the "
35 PRINT"screen.) The right button changes youfrom drawing to erase modes and b
ack (also printed on the screen) while pushing both at once colours in the
space you are in .";
36 PRINT" (Watch with this ifthere is a hole in the outline, the whole screen
will be coloured in)":PRINT
37 PRINT" Press the {SPACE BAR} to start again":PRINT:PRINT
38 PRINT " PRESS JOY STICK BUTTON TO START"
39 IFSTRIG(1)=0THEN39
40 CLS:PRINT:PRINT:PRINT"BLACK OR WHITE BACKGROUND COLOUR?"
41 IFINKEY$=""THEN41
42 IFINKEY$="B"THENBC=1:DC=15:DC$="White":GOTO45
43 IFINKEY$="W"THENBC=15:DC=1:DC$="Black":GOTO45
44 GOTO40
45 SCREEN2,2:CLS:COLORDC,BC,(0,0)-(255,191),BC:COLORDC,DC,(0,0)-(20,191),DC
46 LINE(255,21)-(22,21),DC:LINE-(22,191)
47 COLORDC:CURSOR27,1:PRINT"COLOUR: ":CURSOR27,11:PRINT"MODE: "
48 CURSOR70,1:PRINTDC$:CURSOR70,11:PRINT"Drawing"
49 COLOR1,4,(150,0)-(255,20)
50 CURSOR155,1:PRINT"COPYRIGHT"
51 CURSOR155,11:PRINT"IAN NICHOLSON"
52 C=DC:HH=1:X=135:Y=100:PSET(X,Y),C
53 SCREEN2,2:RETURN
54 IFX=23THENX=24
55 IFX=256THENX=255
56 IFY=21THENY=22
57 IFY=192THENY=191
58 IFHH=1THEN20
59 GOTO21
```

3D Maze by S. Coupe

At last a 3D maze program is available. The program sets you in a "Rat's Eye View" of a complex maze. The object is to find gold and treasure and to murder various nasties. In other words, good wholesome fun!
 To move about the maze, use the arrow cursor keys.



```

1 REM ****3D MAZE****
2 REM (C) S.COUPÉ 1984
5 DIMA$(20)
6 GOTO8000
10 SCREEN2,2:CLS
20 COLOR1,10,(0,0)-(255,191),4
30 COLOR1,15,(16,16)-(108,112),4
40 COLOR1,15,(140,16)-(240,180),4
50 LINE(15,15)-(113,113),1,B:LINE(14,14)-(114,114),1,B
55 LINE(0,0)-(255,191),1,B
60 LINE(135,15)-(248,181),1,B:LINE(134,14)-(249,182),1,B
70 LINE(16,16)-(43,43),7:LINE(16,112)-(43,85),1:LINE(112,16)-(85,43),7:LINE(112,112)-(85,85),1
80 LINE(43,43)-(85,85),1,B:LINE(43,43)-(85,43),7:PAINT(17,16),7
90 COLOR1,10:CURSOR20,150:PRINT"~PLEASE WAIT~":LINE(18,148)-(98,158),1,B:LINE(16,146)-(100,160),1,B
100 FORF=1TO20:READA$(F):NEXT
101 D$="NORTH"
105 MAG1
106 CURSOR140,20:PRINT"LOOKING ";D$:CURSOR140,30:PRINT"TIME 00:00:00"
107 L1=3:EN=50
110 PATTERNS#0,"030F1F3F3F7F7F7F"
120 PATTERNS#1,"E3F1F9FFFFFFFFFB"
130 PATTERNS#2,"C0F0F8FCFCFEFEFE"
140 PATTERNS#3,"C78F9FFFFFFFFFDF"
150 PATTERNS#4,"E9E0C0E0E0F4F5FF"
160 PATTERNS#5,"FFFFFFFCECECE00"
170 PATTERNS#6,"97070307072FAFFF"
180 PATTERNS#7,"FFFFFFFF73737300"
190 PATTERNS#8,"00070F0F1F1F1E3C"
200 PATTERNS#9,"3C38387C7C540000"
210 PATTERNS#10,"0000000000FFFFFF"
220 PATTERNS#11,"FFFFFFFFFFFF0000"
230 PATTERNS#12,"0000000000FFFFFF"
240 PATTERNS#13,"FFFFFFFFFFFF0000"
250 PATTERNS#14,"00E0F0F0FBFB783C"
260 PATTERNS#15,"3C1C1C3E3E2A0000"
270 PATTERNS#16,"00000E17130A250B"
280 PATTERNS#17,"00A502013F000000"
290 PATTERNS#18,"00000000B08080C0"
    
```

```

300 PATTERNS#19,"C0C0C0B000000000"
310 PATTERNS#20,"0000102B6C745AF4"
320 PATTERNS#21,"FF5AFD7E00000000"
325 PATTERNS#25,"FFFFFFFFFFFFFFFF"
326 PATTERNS#28,"000103070F1F3F7F"
327 PATTERNS#30,"FFFEFCF8F0E0C0B0"
329 PATTERNS#34,"000103070F1F3F7F"
330 X=INT(RND(1)*18)+1:Y=INT(RND(1)*28)+1:IF MID$(A$(X),Y,1)<>" " THEN 330
331 PATTERNS#35,"FEFCF8F0E0C0B000"
332 PATTERNS#36,"0000000000000103"
333 PATTERNS#37,"0707000000000000"
334 PATTERNS#38,"01061E3C78F0E0C0"
335 PATTERNS#39,"8000000000000000"
336 PATTERNS#40,"00000000000000060"
337 PATTERNS#41,"F078387FFFFBE000"
338 PATTERNS#43,"0000000080800000"
340 LINE(16,112)-(43,85),1
400 CURSOR140,50:PRINT"Jewels:"CURSOR140,60:PRINT"Energy:50":CURSOR140,70:PRINT
" Swords:"
410 CURSOR140,90:PRINT"Lives:3":CURSOR140,80:PRINT"Nuggets:"
490 TIME$="00:00:00"
500 BLINE(10,130)-(132,190),,BF
510 F$=MID$(A$(X),Y-1,1)
511 IFD$="SOUTH"THENF$=MID$(A$(X),Y+1,1)
512 IFD$="EAST"THENF$=MID$(A$(X-1),Y,1)
513 IFD$="WEST"THENF$=MID$(A$(X+1),Y,1)
514 IFF$=" "THEN1000
515 IFF$="X"THEN1100
520 BLINE(26,46)-(37,100),0,BF
530 LINE(16,112)-(43,85),1
540 F$=MID$(A$(X),Y+1,1)
541 IFD$="SOUTH"THENF$=MID$(A$(X),Y-1,1)
542 IFD$="EAST"THENF$=MID$(A$(X+1),Y,1)
543 IFD$="WEST"THENF$=MID$(A$(X-1),Y,1)
544 IFF$=" "THEN1200
545 IFF$="X"THEN1300
550 BLINE(101,46)-(91,101),,BF
560 LINE(112,112)-(85,85),1
570 F$=MID$(A$(X-1),Y,1)
580 IFD$="SOUTH"THENF$=MID$(A$(X+1),Y,1)
590 IFD$="EAST"THENF$=MID$(A$(X),Y+1,1)
600 IFD$="WEST"THENF$=MID$(A$(X),Y-1,1)
610 IFF$=" "THEN1400
620 IFF$="X"THEN1500
630 BLINE(54,56)-(74,85),,BF
640 LINE(54,85)-(74,85),1
650 IFMID$(A$(X),Y,1)="V"THENGOSUB2000
655 BEEP
656 BLINE(168,30)-(230,40),,BF:CURSOR168,30:PRINTTIME$:T$=TIME$
660 I$=INKEY$
665 IFTIME$<>T$THEN656
670 IFI$=CHR$(28)THEN1600
680 IFI$=CHR$(29)THEN1700
690 IFI$=CHR$(30)THEN1800
695 GOTO660
700 FORA=15TO0STEP-1:SOUND1,310,A:SOUND1,360,A:SOUND1,410,A:NEXT
705 FORF=0TO10:SPRITEF,(0,0),0,0:NEXT
706 BLINE(140,60)-(220,68),,BF:CURSOR140,60:EN=EN-1:PRINT"Energy:";EN
707 IFEN<1THEN5000
710 GOTO500
1000 BLINE(26,101)-(36,92)
1005 BLINE(29,69)-(31,68)
1010 LINE(26,101)-(26,46),1:LINE(36,91)-(34,51),1,BF
1020 LINE(26,46)-(36,51),1:LINE(26,47)-(36,52),1
1030 LINE(26,85)-(36,85),1
1040 GOTO540
1100 BLINE(26,47)-(36,52):LINE(26,46)-(36,51),1
1101 BLINE(26,85)-(36,85)
1102 LINE(16,112)-(43,85),1
1105 LINE(29,69)-(31,68)
1110 LINE(26,101)-(26,46),1:BLINE(36,91)-(34,51),,BF:LINE(36,91)-(36,51),1
1120 GOTO540

```

```

1200 BLINE (101,100)-(91,91)
1205 BLINE (97,69)-(94,68)
1210 LINE (101,101)-(101,46),1:LINE (91,90)-(93,51),1,BF
1220 LINE (101,46)-(91,53),1:LINE (101,47)-(91,54),1
1230 LINE (91,85)-(101,85),1
1240 GOTO570
1300 BLINE (91,90)-(93,51),,BF:BLINE (101,47)-(91,54)
1305 BLINE (91,85)-(101,85):BLINE (101,46)-(101,101)
1310 LINE (101,46)-(101,101),1:LINE (91,90)-(91,51),1
1320 LINE (97,69)-(94,68),1
1330 LINE (101,46)-(91,51),1
1335 LINE (112,112)-(85,85),1
1340 GOTO570
1400 BLINE (57,66)-(59,66)
1410 LINE (54,56)-(74,85),1,B
1420 BLINE (54,85)-(74,85)
1430 GOTO650
1500 LINE (54,56)-(74,85),1,B
1510 LINE (57,66)-(59,66),1
1520 GOTO650
1600 F$=MID$(A$(X),Y+1,1)
1605 IFD$="SOUTH"THENF$=MID$(A$(X),Y-1,1)
1610 IFD$="EAST"THENF$=MID$(A$(X+1),Y,1)
1620 IF D$="WEST"THENF$=MID$(A$(X-1),Y,1)
1630 IFF$="v"THEN 660
1640 IFD$="NORTH"THENY=Y+1:D$="EAST":GOTO1680
1650 IFD$="SOUTH"THENY=Y-1:D$="WEST":GOTO1680
1660 IFD$="EAST"THENX=X+1:D$="SOUTH":GOTO1680
1670 IFD$="WEST"THENX=X-1:D$="NORTH"
1680 BLINE (140,20)-(240,28),,BF:CURSOR140,20:PRINT"LOOKING ";D$
1690 GOTO700
1700 F$=MID$(A$(X),Y-1,1)
1710 IFD$="SOUTH"THENF$=MID$(A$(X),Y+1,1)
1720 IFD$="EAST"THENF$=MID$(A$(X-1),Y,1)
1730 IFD$="WEST"THENF$=MID$(A$(X+1),Y,1)
1740 IFF$="v"THENGOTO660
1750 IFD$="NORTH"THENY=Y-1:D$="WEST":GOTO1680
1760 IFD$="SOUTH"THENY=Y+1:D$="EAST":GOTO1680
1770 IFD$="EAST"THENX=X-1:D$="NORTH":GOTO1680
1780 IFD$="WEST"THENX=X+1:D$="SOUTH":GOTO1680
1790 GOTO700
1800 F$=MID$(A$(X-1),Y,1)
1810 IFD$="SOUTH"THENF$=MID$(A$(X+1),Y,1)
1820 IFD$="EAST"THENF$=MID$(A$(X),Y+1,1)
1830 IFD$="WEST"THENF$=MID$(A$(X),Y-1,1)
1840 IFF$="v"THEN660
1850 IFD$="NORTH"THENX=X-1
1860 IFD$="SOUTH"THENX=X+1
1870 IFD$="EAST"THENY=Y+1
1880 IFD$="WEST"THENY=Y-1
1890 GOTO700
2000 R=INT(RND(1)*50)
2010 IFRND(1)<.2ANDTIME$>"00:04:59"THEN4000
2020 IFR=0THEN2200
2030 IFR=1THEN2400
2040 IFR<10THEN2600
2050 IFR<20THEN2800
2060 IFR<40THEN2900
2070 SPRITE0,(62,90),16,6:SPRITE1,(62,90),20,9
2080 TR=INT(RND(1)*50)+75
2090 GP=GP+TR:CURSOR20,150:PRINT"You have found":PRINTTR;" jewels."
2095 BLINE (140,50)-(240,58),,BF
2100 CURSOR140,50:PRINT"Jewels:";GP
2110 A$(X)=LEFT$(A$(X),Y-1)+" "+RIGHT$(A$(X),30-Y)
2120 RETURN
2200 CURSOR20,150:PRINT"The room is empty."
2210 FORF=1TO500:NEXT:GOTO 2110
2400 CURSOR20,150:PRINT"Sorry,there is
re."
2410 GOTO2210
2600 CURSOR20,150:PRINT"You have found a":CURSOR50,158:PRINT"sword."
nothing he

```

```

2605 SPRITE0,(62,92),36,7:SPRITE1,(62,92),40,1
2610 SW=SW+1:BLINE(140,70)-(220,78),,BF:CURSOR140,70:PRINT"Swords:";SW
2650 GOTO 2110
2800 SCREEN1,1:CLS
2805 PRINT"You have found a man who is selling swords,nuggets and energy pills
."
2806 PRINT:PRINT"***YOU HAVE";GP;" JEWELS***":PRINT
2810 PRINT"1...A sword costs 100 jewels          2...A nugget costs 500 jewels
      3...An energy pill costs 50 jewels"
2815 PRINT:PRINT"DO YOU WANT TO BUY ANYTHING?(Y/N)"
2820 INPUTA$:IFA$<>"Y"THENSREEN2,2:GOTO2940
2830 PRINT"Input number 1 to 3 for the item you want to buy."
2835 INPUTW1:IFW1<1ORW1>3THEN2835
2840 IFW1=1ANDGP>99THENSW=SW+1:GP=GP-100:GOTO2870
2845 IFW1=2ANDGP>499THENG1=G1+1:GP=GP-500:GOTO2870
2850 IFW1=3ANDGP>49THENEN=EN+20:GP=GP-50:GOTO2870
2860 PRINT"I am sorry to say you can't afford it.*****PRESS ANY KEY TO PLAY***
****"
2861 IFINKEY$<>" "THEN2861
2865 IFINKEY$=" "THEN2865
2867 SCREEN2,2:GOTO2940
2870 SCREEN2,2:BLINE(140,60)-(240,68),,BF:CURSOR140,60:PRINT"Energy:";EN
2875 BLINE(140,70)-(240,78),,BF:CURSOR140,70:PRINT"Swords:";SW
2880 BLINE(140,80)-(240,88),,BF:CURSOR140,80:PRINT"Nuggets:";G1
2885 BLINE(140,50)-(240,58),,BF:CURSOR140,50:PRINT"Jewels:";GP
2890 GOTO2940
2900 SPRITE0,(62,92),24,5:SPRITE1,(62,92),28,7:SPRITE2,(62,92),32,4
2910 CURSOR20,150:COLOR1:PRINT"Are you going to                               open th
e box ?"
2920 IFINKEY$="Y"THEN3000
2930 IFINKEY$<>"N"THEN2920
2940 BEEP:BLINE(20,150)-(130,170),,BF:GOTO2110
3000 R=INT(RND(1)*5)+1
3001 A$(X)=LEFT$(A$(X),Y-1)+" "+RIGHT$(A$(X),30-Y)
3005 BLINE(20,150)-(130,170),,BF
3010 ONRGOTO3020,3100,3200,3300
3020 FORF=0TO3:SPRITEF,(0,0),0,0:NEXT
3025 OUT127,228:FORF=240TO255:OUT127,F:FORU=0TO15:NEXT:NEXT
3030 CURSOR40,150:PRINT"It exploded!":CURSOR20,160:PRINT"You loose a life."
3040 L1=L1-1
3045 FORU=1TO4
3050 FORF=1000TO120STEP-30:SOUND1,F,8:NEXT:NEXT
3055 BLINE(140,90)-(220,98),,BF:CURSOR140,90:PRINT"LIVES:";L1
3056 IFL1=0THEN5000
3060 GOTO2940
3100 CURSOR20,150:PRINT"You have found a":CURSOR20,158:PRINT" gold nugget."
3110 G1=G1+1
3115 BLINE(140,80)-(240,88),,BF:CURSOR140,80:PRINT"Nuggets:";G1
3120 FORF=0TO500:NEXT
3130 GOTO2940
3200 CURSOR20,150:PRINT"You have found some ":CURSOR20,158:PRINT"energy choclate
."
3210 EN=EN+10
3215 BLINE(140,60)-(240,68),,BF:CURSOR140,60:PRINT"Energy:";EN
3220 FORF=0TO500:NEXT:GOTO2940
3300 CURSOR20,150:PRINT"It's empty"
3310 FORF=0TO100:BEEP1:BEEPO:NEXT
3320 GOTO2940
4000 SPRITE0,(55,74),0,8:SPRITE1,(55,90),4,8
4010 SPRITE2,(47,84),8,12:SPRITE3,(63,84),12,12
4020 FORF=1TO70:SOUND1,2000,15:SOUND0:NEXT
4030 IFSW<3THEN4200
4040 CURSOR20,150:PRINT"You killed him"
4045 SW=SW-3:BLINE(140,70)-(240,78),,BF:CURSOR140,70:PRINT"Swords:";SW
4050 FORF=1TO500:NEXT:GOTO2940
4200 CURSOR20,150:PRINT"He killed you!"
4210 FORF=0TO500:NEXT
5000 SCREEN1,1:CLS
5010 SC=SC+(SW*10)+GP+(G1*100)+(EN*5)
5020 PRINT"Tough luck,you have lost all of your lives.Better luck next time.At
least you managed to score";SC

```


SEGA NEW ZEALAND . . . IS CROSSING THE TASMAN

Grandstand Computer Ltd has now passed its support for SEGA to SEGA SOFTWARE SUPPORT . . . that's us. We in New Zealand have rested on our laurels far too long and now we're crossing the Tasman bringing

- our renowned user magazine SEGA COMPUTER
- all our famous Kiwi software



Now you will be able to subscribe direct to SEGA COMPUTER magazine or simply collect it direct from your local Sega user group. We are also looking throughout Australia for quality contributions of software programs, user tips, reviews and comments to include in our magazine . . . And for Australian professional software writers, we are able to market your software here to our list of almost 4000 SEGA owners. And the list is growing daily. Just send us your name and address and we'll send you a sample magazine and our catalogue — FREE OF CHARGE.

We look forward to hearing from you. You'll certainly hear more from us.

SEGA SOFTWARE SUPPORT, P.O. BOX 10-207, Auckland, 4, New Zealand. Phone NZ (9) 603-274

```

350 CIRCLE (30,130),4,5,,BF
360 CIRCLE(14,40),70, 9,1.3,.95,.19
370 LINE (65,145)-(93,145),1
380 LINE (65,145)-(70,135),1
390 LINE (70,135)-(75,131),1
400 LINE (75,131)-(79,129),1
410 LINE (79,129)-(85,135),1
420 LINE (85,135)-(93,145),1
430 PAINT(80,139),1
440 CIRCLE(80,150),29,5,.3,.85,.65
450 CIRCLE(80,149),22,5,.2,.80,.70
460 CIRCLE(26,25),10,11,,BF
470 LINE (8,8)-(120,176),1,B
480 PSET (80,150),6
490 PSET (81,149),6
500 PSET (81,150),6
510 PSET (82,150),6
520 CURSOR 130,20:COLOR1:PRINT CHR$(17
);" P I A N O "
530 CURSOR 142,35:PRINT " T U N E "
540 PRINT CHR$(16)
550 RESTORE 590
560 FOR J=1 TO 54:READ F:SOUND1,F,14:S
OUND2,F+2,14:SOUND3,F*2,14:FOR DE=1 TO
30:NEXT DE,J
570 READ D
580 IF D=0 THEN SOUND 0
590 DATA 330,294,262,330,294,262,392,3
49,330,392,349,330,262,330,262,330,262
,392,262,392
600 DATA 392,262,392,262,330,262,330,2
62,330,523,494,440,523,494,440
610 DATA 659,587,659,523,587,494,523,4
40,494,392,440,349,392,330,349,294,330
,262,262,0
620 CURSOR 8,180:PRINT "Press a key t
o begin"
630 IF INKEY$="" THEN 630
640 REM PIANO
650 CLS
660 PRINT "      2   4 5   7 8 9   - ^
"
670 PRINT "
"
680 PRINT "
"
690 PRINT " 'NOTE' AMENDMENT AT END OF LISTING.
"
700 PRINT "
"
710 PRINT "
"
720 PRINT
730 PRINT "      Q W E R T Y U I O P @ [
"
740 PRINT "      -C-"

```

```

750 PRINT :PRINT "      Press keys to ma
ke like a      'PIANO'
"
760 A=1
770 A$=INKEY$
780 SOUND0
790 IF A$="Q" THEN 1000
800 IF A$="2" THEN 1030
810 IF A$="W" THEN 1060
820 IF A$="E" THEN 1090
830 IF A$="4" THEN 1120
840 IF A$="R" THEN 1150
850 IF A$="5" THEN 1180
860 IF A$="T" THEN 1210
870 IF A$="Y" THEN 1240
880 IF A$="7" THEN 1270
890 IF A$="U" THEN 1300
900 IF A$="8" THEN 1330
910 IF A$="I" THEN 1360
920 IF A$="9" THEN 1390
930 IF A$="O" THEN 1420
940 IF A$="P" THEN 1450
950 IF A$="-" THEN 1480
960 IF A$="@ " THEN 1510
970 IF A$="^" THEN 1540
980 IF A$="[" THEN 1570
990 GOTO 770
1000 SOUND A,220,15:SOUND 2,220+10,8
1010 IF INKEY$<>" " THEN 1000
1020 GOTO 770
1030 SOUND A,233,15:SOUND2,233+10,8
1040 IF INKEY$<>" " THEN 1030
1050 GOTO 770
1060 SOUND A,247,15:SOUND2,247+10,8
1070 IF INKEY$<>" " THEN 1060
1080 GOTO 770
1090 SOUND A,262,15:SOUND2,262+10,8
1100 IF INKEY$<>" " THEN 1090
1110 GOTO 770
1120 SOUND A,277,15:SOUND2,277+10,8
1130 IF INKEY$<>" " THEN 1120
1140 GOTO 770
1150 SOUND A,294,15:SOUND2,294+10,8
1160 IF INKEY$<>" " THEN 1150
1170 GOTO 770
1180 SOUND A,311,15:SOUND2,311+10,8
1190 IF INKEY$<>" " THEN 1180
1200 GOTO 770
1210 SOUND A,330,15 :SOUND2,330+10,8
1220 IF INKEY$<>" " THEN 1210
1230 GOTO 770
1240 SOUND A,349,15:SOUND2,349+10,8
1250 IF INKEY$<>" " THEN 1240
1260 GOTO 770
1270 SOUND A,370,15:SOUND2,370+10,8
1280 IF INKEY$<>" " THEN 1270

```



```

1290 GOTO 770
1300 SOUND A,392,15 :SOUND2,392+10,8
1310 IF INKEY$("<>") THEN 1300
1320 GOTO 770
1330 SOUND A,415,15 :SOUND2,415+10,8
1340 IF INKEY$("<>") THEN 1330
1350 GOTO 770
1360 SOUND A,440,15 :SOUND2,440+10,8
1370 IF INKEY$("<>") THEN 1360
1380 GOTO 770
1390 SOUND A,466,15 :SOUND2,466+10,8
1400 IF INKEY$("<>") THEN 1390
1410 GOTO 770
1420 SOUND A,494,15 :SOUND2,494+10,8
1430 IF INKEY$("<>") THEN 1420
1440 GOTO 770
1450 SOUND A,523,15 :SOUND2,523+10,8
1460 IF INKEY$("<>") THEN 1450
1470 GOTO 770
1480 SOUND A,554,15 :SOUND2,554+10,8
1490 IF INKEY$("<>") THEN 1480
1500 GOTO 770
1510 SOUND A,587,15 :SOUN2,587+10,8
1520 IF INKEY$("<>") THEN 1510
1530 GOTO 770
1540 SOUND A,622,15 :SOUND2,622+10,8
1550 IF INKEY$("<>") THEN 1540
1560 GOTO 770
1570 SOUND A,659,15 :SOUND2,659+10,8
1580 IF INKEY$("<>") THEN 1570
1590 GOTO 770

```

```

660 PRINT "      2  4 5  7 8 9  - ^
"
670 PRINT "      U  U U  U U U  U U
"
680 PRINT "      U  U U  U U U  U U
"
690 PRINT "      U U U U U U U U U U U
"
700 PRINT "      U U U U U U U U U U U
"
710 PRINT "      U U U U U U U U U U U
"
720 PRINT
730 PRINT "      Q W E R T Y U I O P @ [
"

```

732 REM Lines 670-710 are with the GRA
PH key to give the piano keys.

LOCAL SEGA USERS CONTACTS

AUCKLAND CENTRAL SEGA USERS CLUB
(see next issue)

NAPIER SEGA USERS CLUB
Sec E.P. Lins
41 Higgins Street
NAPIER

SOUTH COROMANDEL USERS CLUB
P.O. Box 183
WHANGAMATA
Contact: Sid
Ph. 58-775 Whangamata

B.O.P SEGA USERS CLUB
Sega 102B Hinewa Road
TAURANGA
Sec P.J. McDonald
Ph 64826 a/h
Tauranga

PUKEKOHE SEGA USERS CLUB
C/o 4 Roose Avenue
Pukekohe
Contact: Selwyn
Ph. Pukekohe 86-583

SOUTH TARANAKI MICROCOMPUTER SOCIETY
D.M. Beale
7A Clive Street
HAWERA

CHRISTCHURCH USER'S CLUB
Contact Graham Rudman
29 Primrose Street
CHRISTCHURCH 5

ROTORUA SEGA USERS CLUB
Contact: Terry Cole
Ph. 59-325 a.h.

WELLINGTON USERS CLUB
Contact: Shaun Parsons
P.O. Box 1871
Postal Central
Wellington
Ph 897-095 a/h
Ph 727-666 work

GISBORNE AREA USER'S CLUB
Trevor Gardiner
Ph. 83-068 HM
or 87-175 WK

TOKOROA SEGA USERS GROUP
C/o 1 Pio Pio Place
TOKOROA
Contact Geoff Phone Number 67-105
Tokoroa

The above are contact names and addresses for
Sega Users Clubs. If you wish to have your club
advertised write to Sega Users Club P.O. Box
10-207, Auckland, 4.

HAMILTON SEGA USER'S CLUB
P.O. Box 1548 Hamilton
President: Mr Colin Bell 73-826
Secretary: Mrs Anne Thrush Ph 437-312
Meetings held fortnightly on Monday at Waikato
Technical Institute in Room F129.

SOUTH CANTERBURY USERS CLUB
P.O. Box 73 Timaru New Zealand
President: Geoff McCayghan
Secretary: Lloyd Van Der Krogt
Contact Phs: 60-756, 61-412 TIMARU

If you have set up a local area user's club and you
would like us to publish the details concerning
your club please send them to us and we will
publish the information for no charge.